
Chapitre 1

Première prise en main de Maxima

Maxima (http ://maxima.sourceforge.net/) est un logiciel de calcul formel, descendant du logiciel
Macsyma développé dès 1968 au MIT (institut de recherche et université américaine). Il est dispo-
nible sous linux, Mac OS X et Windows. C’est un logiciel libre distribué sous licence GNU GPL,
programmé dans le langage Lisp.

Ce logiciel possède une interface graphique qui est wxmaxima. C’est cette interface graphique avec
maxima que l’on utilise ici.

Le logiciel wxmaxima se présente graphiquement en deux parties : la zone de menu et la zone
d’exécution des commandes. La zone de menu, comme beaucoup de logiciels comporte les sous
menus « Fichiers », « Edition » et « Aide ». Dans la zone d’exécution des commandes, on peut
écrire du texte comme ce que vous êtes en train de lire et aussi bien sûr exécuter des commandes !

Les commandes sont toujours précédées d’une flêche lorsqu’elles sont vides ou de %i et un nombre
entouré par des parenthèses lorsquelles contiennent une commande qui a été exécutée. Si la com-
mande (ci-dessous) n’a pas été exécutée, il faut mettre le curseur dessus et appuyer simultanément
les touches Majuscule et entrée (Maj+Entrée).

(%i1) 2+5;

(%o1) 7

La lettre i signifie input (entrée) et le o output (sortie). Le pourcentage sert à dire que i1 est un
nom, c’est le nom de la première entrée. On peut les utiliser dans la suite des commandes. Un
poucentage seul est le nom du dernier résultat obtenu.

(%i2) %;

(%o2) 7

(%i3) %o1;

(%o3) 7

(%i4) %i1;

(%o4) 7

Le point virgule « ; » marque la fin de la commande à éxécuter. À la place du point virgule, on
peut mettre un dollar pour ne pas afficher le résultat. Cela peut être utile si l’on ne souhaite pas
faire apparaître de résultats intermédiaires.

(%i5) 2+5$

7



8 CHAPITRE 1. PREMIÈRE PRISE EN MAIN DE MAXIMA

(%i6) %i5+2;

(%o6) 9

On peut aussi écrire plusieurs commandes dans une même cellule :

(%i7) 2+7;
7/8;

(%o7) 9

(%o8)
7

8

Si on éxécute plusieurs fois une commande, la numérotation augmente et certains numéros dis-
paraissent. On peut remettre à jour la numérotation en utilisant "Menu/Maxima/Redémarrer
Maxima" puis "Menu/Cell/Réévaluer toutes les cellules". Avec le clavier on peut utiliser : Alt+M+R
puis CTRL+R.



Chapitre 2

Les nombres

2.1 Opérations sur les nombres

2.1.1 Somme et différence

(%i1) 2+3;
2-3;

(%o1) 5

(%o2) − 1

2.1.2 Produit et division

(%i3) 5*45;

(%o3) 225

(%i4) -63/49;

(%o4) − 9

7

2.1.3 Arrondis

Si on veut un arrondi, on utilise la commande float :

(%i5) float(%);

(%o5) − 1.285714285714286

On peut l’avoir directement si on écrit un des nombres sous forme décimale :

(%i6) -63.0/49;
-63/49.0;

(%o6) − 1.285714285714286

(%o7) − 1.285714285714286

9

2.2 Des fonctions sur les entiers

2.2.1 Factorisation

(%i8) factor(6);
factor(45454566780);

(%o8) 2 3

(%o9) 22 32 5 72 19 271241

2.2.2 Racine carrée

La racine est un entier

(%i10) sqrt(4);

(%o10) 2

La racine est non entière et le résultat donne la racine carrée :

(%i11) sqrt(2);

(%o11)
√

2

Les racine est non entière et on obtient une valeur approchée :

(%i12) sqrt(2.0);

(%o12) 1.414213562373095

Valeur exacte écrite sous forme de puissance

(%i13) sqrt(8);

(%o13) 2
3
2

Valeur approchée

(%i14) sqrt(8.0);

(%o14) 2.82842712474619

2.2.3 Puissances

(%i15) 2^3;

(%o15) 8

(%i16) 45^3;

(%o16) 91125

2.2.4 Valeur absolue

Valeur absolue :

(%i17) abs(-4);

(%o17) 4

2.2.5 Factorielle

La factorielle d’un nombre

(%i18) 3!;

(%o18) 6

(%i19) 50!;

(%o19) 30414093201713378043612608166064768844377641568960512000000000000

2.3 Représentation des nombres

La notation scientifique

(%i20) sqrt(23.0)^2-23;

(%o20) − 3.5527136788005009 10−15

Si l’on souhaite avoir d’avantage de précision, on indique la précision souhaitée, par exemple avec
100 chiffres après la décimale :

(%i21) fpprec:100;

(%o21) 100

On utilise ensuite la commande bfloat :

(%i22) bfloat(sqrt(4700000));

(%o22) 2.1679483388678799418989624480[43digits]4185379293026850227472580943b3

où b3 signifie que le résultat est multiplié par 10^3 (il y a 43 chiffres qui ne sont pas affichés, ils
sont remplacés par [43 digits]). On peut aussi écrire la commande en une seule ligne :

(%i23) sqrt(8), bfloat, fpprec:50;

(%o23) 2.8284271247461900976033774484193961571393437507539b0

Chapitre 3

Les variables

3.1 Les variables

Déclaration d’une variable

(%i1) a:2;

(%o1) 2

Maintenant la variable a, contient la valeur 2 :

(%i2) a;

(%o2) 2

Si l’on souhaite désaffecter la variable a :

(%i3) kill(a);

(%o3) done

(%i4) a;

(%o4) a

On peut développer :

(%i5) expand((a+1)^2);

(%o5) a2 + 2 a+ 1

On peut aussi factoriser :

(%i6) factor(a^2+5*a+6);

(%o6) (a+ 2) (a+ 3)

Les identités remarquables avec la commande expand :

(%i7) expand((a+b)^2);

(%o7) b2 + 2 a b+ a2

(%i8) expand((a-b)^2);

(%o8) b2 − 2 a b+ a2

13

(%i9) expand((a-b)*(a+b));

(%o9) a2 − b2

Les identités remarquables avec la commande factor :

(%i10) factor(a^2+2*a*b+b^2);

(%o10) (b+ a)
2

(%i11) factor(a^2-2*a*b+b^2);

(%o11) (b− a)
2

(%i12) factor(a^2-b^2);

(%o12) − (b− a) (b+ a)

Après utilisation de declare(a,mainvar) ; afin d’avoir la variable a en premier :

(%i13) declare(a,mainvar);

(%o13) done

(%i14) expand((a+b)^2);
expand((a-b)^2);
expand((a-b)*(a+b));

(%o14) a2 + 2 b a+ b2

(%o15) a2 − 2 b a+ b2

(%o16) a2 − b2

(%i17) factor(a^2+2*a*b+b^2);
factor(a^2-2*a*b+b^2);
factor(a^2-b^2);

(%o17) (a+ b)
2

(%o18) (a− b)2

(%o19) (a− b) (a+ b)

3.2 Les constantes

Le nombre PI

(%i20) %pi;

(%o20) π

(%i21) float(%pi);

(%o21) 3.141592653589793

Le nombre e



3.3. SUBSTITUTIONS DE VARIABLES 15

(%i22) %e;

(%o22) e

(%i23) float(%e);

(%o23) 2.718281828459045

3.3 Substitutions de variables

Il peut parfois être utilise de remplacer une variable par une expression ou par une valeur. La
fonction subst permet de réaliser cette action :

(%i24) subst(x^2, a,(a+1)^2);

(%o24)
(
x2 + 1

)2
(%i25) subst(2, a,(a+1)^2);

(%o25) 9

Chapitre 4

Les équations

4.1 Les équations

4.1.1 Définition d’une équation

On commence par définir une équation et on lui donne un nom :

(%i1) equation1:x^2-1=0;

(%o1) x2 − 1 = 0

4.1.2 Résolution exacte d’une équation

On résout l’équation suivante, où l’inconnue est x (dans le menu : Menu/Equations/Résoudre...) :

(%i2) solve(equation1,x);

(%o2) [x = −1, x = 1]

On peut aussi l’écrire directement :

(%i3) solve(x^2-1=0,x);

(%o3) [x = −1, x = 1]

4.1.3 Utilisation des solutions obtenues

Les résultats précédents sont donnés sous forme d’équation x = . . ..

Pour obtenir la première équation, on écrit :

(%i4) solve(x^2-1=0,x)[1];

(%o4) x = −1

Pour la deuxième :

(%i5) solve(x^2-1=0,x)[2];

(%o5) x = 1

Pour récupérer le membre de droite, on utilise rhs (right hand side) :

17

(%i6) rhs(solve(x^2-1=0,x)[1]);
rhs(solve(x^2-1=0,x)[2]);

(%o6) − 1

(%o7) 1

Si on souhaite récupérer la variable, on utilise lhs (left hand side) :

(%i8) lhs(solve(x^2-1=0,x)[1]);
lhs(solve(x^2-1=0,x)[2]);

(%o8) x

(%o9) x

Pour vérifier qu’un nombre annule une expression :

(%i10) subst(-1, x, x^2-1);

(%o10) 0

4.1.4 Résolution approchée d’une équation

Si une solution exacte de l’équation n’existe pas, on obtient l’équation :

(%i11) solve(x^7-2*x+5=0,x);

(%o11) [0 = x7 − 2x+ 5]

Dans ce cas on utilise une méthode de résolution numérique, il faut donner un intervalle où la
fonction change une seule fois de signe.

(%i12) find_root(x^7-2*x+5=0, x, -2, 0);

(%o12) − 1.337973214895086

En changeant l’intervalle, on peut obtenir toutes les solutions :

(%i13) find_root(x^5+20*x^2+2*x-1=0, x, -10, 10);
find_root(x^5+20*x^2+2*x-1=0, x, -10, -1);
find_root(x^5+20*x^2+2*x-1=0, x, -1, 0);

(%o13) 0.17910867241289

(%o14) − 2.673637481699788

(%o15) − 0.27931420183824

4.1.5 Résolution d’équations polynomiales

Si l’équation est polynomiale, on dispose de realroots :

(%i16) realroots(x^5+20*x^2+2*x-1);

(%o16) [x = −89712387

33554432
, x = − 9372229

33554432
, x =

6009889

33554432
]

Pour obtenir les solutions réelles approchées, on utilise float :

(%i17) float(realroots(x^5+20*x^2+2*x-1));

(%o17) [x = −2.673637479543686, x = −0.27931419014931, x = 0.17910864949226]

Par contre solve ne trouve pas ces solutions !

(%i18) solve(x^5+20*x^2+2*x-1=0, x);

(%o18) [0 = x5 + 20x2 + 2x− 1]

4.2 Les inéquations

Pour les inéquations, il existe le package solve_rat_ineq qui fonctionne pour les expressions ra-
tionnelles.

(%i19) load(solve_rat_ineq)$

(%i20) solve_rat_ineq((2*x+3)*(x)>0);

(%o20) [[x < −3

2
], [x > 0]]

(%i21) solve_rat_ineq((x-1)^2*(x+1)^2>0);

(%o21) [[x < −1], [x > −1, x < 1], [x > 1]]

Chapitre 5

Les fonctions

5.1 Définition d’une fonction

(%i1) f(x):=x^2;

(%o1) f (x) := x2

Avec define :

(%i2) define(g(x), x^3-1);

(%o2) g (x) := x3 − 1

5.2 Tableau de valeurs d’une fonction

On utilise les listes :

(%i3) valeursdex:makelist(-1+k/10,k,0, 20);
imagesdesvaleursdex:map(f,valeursdex);

(%o3) [−1,− 9

10
,−4

5
,− 7

10
,−3

5
,−1

2
,−2

5
,− 3

10
,−1

5
,− 1

10
, 0,

1

10
,

1

5
,

3

10
,

2

5
,

1

2
,

3

5
,

7

10
,

4

5
,

9

10
, 1]

(%o4) [1,
81

100
,

16

25
,

49

100
,

9

25
,

1

4
,

4

25
,

9

100
,

1

25
,

1

100
, 0,

1

100
,

1

25
,

9

100
,

4

25
,

1

4
,

9

25
,

49

100
,

16

25
,

81

100
, 1]

(%i5) wxplot2d([discrete, valeursdex,imagesdesvaleursdex],[style, points]);

(%t5)

21

(%o5)

5.3 Courbe représentative d’une fonction

On donne l’expression de f(x) et l’intervalle sur lequel on souhaite représenter la fonction. L’inter-
valle des coordonnées se calcule automatiquement.

(%i6) wxplot2d(f(x),[x,-1,1]);

(%t6)

(%o6)

On peut aussi donner cet intervalle.

(%i7) wxplot2d(f(x),[x,-1,1], [y, -1,2]);

(%t7)

(%o7)

Pour tracer plusieurs courbes à la fois, on donne la liste des expressions des fonctions à représenter.

(%i8) wxplot2d([f(x),g(x)],[x,-2,2]);

(%t8)

(%o8)

5.4 Dérivée d’une fonction

(%i9) define(f(x),x^20+x^15);
diff(f(x),x);
define(df(x),diff(f(x),x));

(%o9) f (x) := x20 + x15

(%o10) 20x19 + 15x14

(%o11) df (x) := 20x19 + 15x14

Dérivée seconde :

(%i12) diff(f(x),x,2);

(%o12) 380x18 + 210x13

Dérivées d’ordre supérieur :

(%i13) diff(f(x),x,5);

(%o13) 1860480x15 + 360360x10

Remarque : pour définir la dérivée, il faut utiliser define et non := Voici deux exemples illustrant
une différence :

(%i14) g(x):=x^2;
h(x):=x^3;

(%o14) g (x) := x2

(%o15) h (x) := x3

(%i16) dg(x):=diff(g(x),x);
define(dh(x), diff(h(x),x));

(%o16) dg (x) := diff (g (x) , x)

(%o17) dh (x) := 3x2

Jusqu’ici pas de différence, en effet :

(%i18) dg(x);
dh(x);

(%o18) 2x

(%o19) 3x2

mais pour le calcul en une valeur := donne une erreur :

(%i20) dg(2);
dh(2);

diff : secondargumentmustbeavariable; found”2#0” : dg(x = 2) − −anerror.Todebugthistry :
debugmode(true);

(%o21) 12

Lorsque l’on demande dg(2), Maxima exécute la commande diff(g(2), 2) ce qui n’est pas possible
car le deuxième paramètre de diff doit-être une variable. On dérive par rapport à une variable.

5.5 Limites d’une fonction

En plus l’infini

(%i22) limit(f(x),x,inf);

(%o22) ∞

Lorsque le signe n’est pas précisé, c’est plus l’infini. Lorsque la limite est indéfinie et infinie en
valeur absolue, on obtient le mot "infinity" :

(%i23) limit(1/x,x,0);

(%o23) infinity

En 0+ et 0− :

(%i24) limit(1/x, x, 0, plus);

(%o24) ∞

(%i25) limit(1/x, x, 0, minus);

(%o25) −∞

En moins l’infini

(%i26) limit(f(x),x,minf);
limit(f(x),x,-inf);

(%o26) ∞

(%o27) ∞

En une valeur

(%i28) limit(f(x),x,2);

(%o28) 1081344

Limite indéfinie

(%i29) limit(sin(x), x, inf);

(%o29) ind

5.6 Fonctions affines

(%i30) kill(a)$
kill(b)$

(%i32) h(x):=a*x+b;

(%o32) h (x) := a x+ b

Zéro d’une fonction affine

(%i33) solve(h(x)=0,x);

(%o33) [x = − b
a

]

Dérivée d’une fonction affine

(%i34) diff(h(x),x);

(%o34) a

Limites

(%i35) limit(f(x),x,-inf);
limit(f(x),x,x[0]);
limit(f(x),x,inf);

(%o35) ∞

(%o36) x200 + x150

(%o37) ∞

Courbe d’une fonction affine : droite

(%i38) a:1.5$
b:2$
h(x);

(%o40) 1.5x+ 2

(%i41) wxplot2d([h(x)], [x,-5,5])$

(%t41)

5.7 Fonctions du second degré

(%i42) kill(a)$
kill(b)$
kill(c)$

(%i45) f(x):=a*x^2+b*x+c;

(%o45) f (x) := a x2 + b x+ c

Zéros d’un polynôme du second degré

(%i46) solutions:solve(f(x)=0,x);

(%o46) [x = −
√
b2 − 4 a c+ b

2 a
, x =

√
b2 − 4 a c− b

2 a
]

Attention cette expression est valable pour toutes les valeurs réelles de ∆. Si Delta est négatif le
nombre %i apparaît, voir les exemples suivants. Dans le cas où Delta = b^2-4*a*c est positif :

(%i47) subst(1, c,subst(4, b, subst(1, a, solutions)));

(%o47) [x = −2
√

3 + 4

2
, x =

2
√

3− 4

2
]

Dans le cas où Delta = b^2-4*a*c est nul :

(%i48) subst(1, c,subst(2, b, subst(1, a, solutions)));

(%o48) [x = −1, x = −1]

Dans le cas où Delta = b^2-4*a*c est négatif :

(%i49) subst(4, c,subst(2, b, subst(1, a, solutions)));

(%o49) [x = −2
√

3 i+ 2

2
, x =

2
√

3 i− 2

2
]

Dérivée d’une fonction polynôme

(%i50) define(df(x),diff(f(x),x));

(%o50) df (x) := 2 a x+ b

Zéro de la dérivée

(%i51) solve(df(x)=0,x);

(%o51) [x = − b

2 a
]

(%i52) a:1.5$
b:2$
c:-1$
f(x);

(%o55) 1.5x2 + 2x− 1

Limites

(%i56) limit(f(x),x,-inf);
limit(f(x),x,x[0]);
limit(f(x),x,inf);

(%o56) ∞

(%o57) 1.5x20 + 2x0 − 1

(%o58) ∞

Courbe d’une fonction du second degré

(%i59) wxplot2d([f(x)], [x,-5,5])$

(%t59)

5.8 Fonction exponentielle

Définition

(%i60) define(f(x),exp(x));

(%o60) f (x) := ex

Valeurs particulières

(%i61) exp(0);

(%o61) 1

(%i62) exp(1);

(%o62) e

Dérivée

(%i63) define(df(x),diff(f(x),x));

(%o63) df (x) := ex

Limites

(%i64) limit(f(x),x,-inf);
limit(f(x),x,x[0]);
limit(f(x),x,inf);

(%o64) 0

(%o65) ex0

(%o66) ∞

(%i67) limit(exp(x)/x,x,inf);

(%o67) ∞

Courbe

(%i68) wxplot2d([f(x)], [x,-2,2])$

(%t68)

Composée avec une fonction affine

(%i69) kill(a)$
kill(b)$
define(f(x),exp(a*x+b));
define(df(x),diff(f(x),x));
a:-1$ b:2$
wxplot2d([f(x)], [x,-2,2])$

(%o71) f (x) := ea x+b

(%o72) df (x) := a ea x+b

(%t75)

Relations fonctionnelles

(%i76) kill(a)$
kill(b)$

(%i78) exp(a)*exp(b);

(%o78) eb+a

(%i79) 1/exp(a);

(%o79) e−a

(%i80) exp(a)/exp(b);

(%o80) ea−b

(%i81) (exp(a))^b;

(%o81) ea b

(%i82) sqrt(exp(a));

(%o82) e
a
2

5.9 Fonction logarithme népérien

Definition

(%i83) define(f(x),log(x));

(%o83) f (x) := log (x)

Valeurs particulières

(%i84) log(1);
log(1.0);

(%o84) 0

(%o85) 0.0

(%i86) log(%e);
log(2.718);

(%o86) 1

(%o87) 0.99989631572895

Dérivée

(%i88) define(df(x),diff(f(x),x));

(%o88) df (x) :=
1

x

Limites

(%i89) limit(f(x),x,-inf);
limit(f(x),x,3);
limit(f(x),x,inf);

(%o89) infinity

(%o90) log (3)

(%o91) ∞

(%i92) limit(log(x)/x,x,inf);

(%o92) 0

Courbe

(%i93) wxplot2d([f(x)], [x,0.001,5])$

(%t93)

Composée avec une fonction affine

(%i94) kill(a)$
kill(b)$
define(f(x),log(a*x+b));
define(df(x),diff(f(x),x));
a:-1$ b:2$
wxplot2d([f(x)], [x,-2,1.9999])$

(%o96) f (x) := log (a x+ b)

(%o97) df (x) :=
a

a x+ b

(%t100)

Logarithme et exponentielle

(%i101)log(exp(x));

(%o101)x

(%i102)exp(log(x));

(%o102)x

Relations fonctionnelles

(%i103)kill(a)$
kill(b)$

(%i105)logcontract(log(a)+log(b));

(%o105)log (a b)

(%i106)logcontract(-log(a));

(%o106)− log (a)

(%i107)logcontract(log(a)-log(b));

(%o107)log
(a
b

)
(%i108)log(a^b);

(%o108)log (a) b

(%i109)log(sqrt(a));

(%o109)
log (a)

2

5.10 Fonctions trigonométriques

Définition

(%i110)define(f(x),cos(x));
define(g(x),sin(x));

(%o110)f (x) := cos (x)

(%o111)g (x) := sin (x)

Valeurs particulières entre 0 et 2 ∗ π

(%i112)valeursdex:sort(append(makelist(k*%pi/6,k,0, 12),[%pi/4, 3*%pi/4, 5*%pi/4, 7*%pi/4]));
valeursdecosx:map(f,valeursdex);
valeursdesinx:map(g,valeursdex);

(%o112)[0,
π

6
,
π

4
,
π

3
,
π

2
,

2π

3
,

3π

4
,

5π

6
, π,

7π

6
,

5π

4
,

4π

3
,

3π

2
,

5π

3
,

7π

4
,

11π

6
, 2π]

(%o113)[1,
√

3

2
,

1√
2
,

1

2
, 0,−1

2
,− 1√

2
,−
√

3

2
,−1,−

√
3

2
,− 1√

2
,−1

2
, 0,

1

2
,

1√
2
,

√
3

2
, 1]

(%o114)[0,
1

2
,

1√
2
,

√
3

2
, 1,

√
3

2
,

1√
2
,

1

2
, 0,−1

2
,− 1√

2
,−
√

3

2
,−1,−

√
3

2
,− 1√

2
,−1

2
, 0]

Dérivées

(%i115)define(df(x),diff(f(x),x));
define(dg(x),diff(g(x),x));

(%o115)df (x) := −sin (x)

(%o116)dg (x) := cos (x)

Limites

L’affichage de ind veut dire indéfinie, la limite n’existe pas !

(%i117)limit(f(x),x,-inf);
limit(f(x),x,3);
limit(f(x),x,inf);
limit(g(x),x,-inf);
limit(g(x),x,3);
limit(g(x),x,inf);

(%o117)ind

(%o118)cos (3)

(%o119)ind

(%o120)ind

(%o121)sin (3)

(%o122)ind

Courbes

(%i123)wxplot2d([f(x),g(x)], [x,0,2*%pi])$

(%t123)

Composée avec une fonction affine

(%i124)kill(a)$
kill(b)$
define(f(x),cos(a*x+b));
define(df(x),diff(f(x),x));
define(g(x),sin(a*x+b));
define(dg(x),diff(f(x),x));
a:-1$ b:2$
wxplot2d([f(x),g(x)], [x,0,20])$

(%o126)f (x) := cos (a x+ b)

(%o127)df (x) := −a sin (a x+ b)

(%o128)g (x) := sin (a x+ b)

(%o129)dg (x) := −a sin (a x+ b)

(%t132)

Relations

(%i133)kill(a)$
kill(b)$

(%i135)trigexpand(cos(a+b));

(%o135)cos (a) cos (b)− sin (a) sin (b)

(%i136)trigexpand(cos(a-b));

(%o136)sin (a) sin (b) + cos (a) cos (b)

(%i137)trigexpand(sin(a+b));

(%o137)cos (a) sin (b) + sin (a) cos (b)

(%i138)trigexpand(sin(a-b));

(%o138)sin (a) cos (b)− cos (a) sin (b)

5.11 Primitives et intégrales d’une fonction

5.11.1 Primitives

(%i139)kill(all)$

(%i1) integrate(a*x+b, x);

(%o1)
a x2

2
+ b x

(%i2) integrate(a*x^2+b*x+c,x);

(%o2)
a x3

3
+
b x2

2
+ c x

(%i3) integrate(exp(a*x+b),x);

(%o3)
ea x+b

a

(%i4) integrate(log(a*x+b),x);

(%o4)
(a x+ b) log (a x+ b)− a x− b

a

(%i5) integrate(cos(a*x+b),x);

(%o5)
sin (a x+ b)

a

(%i6) integrate(sin(a*x+b),x);

(%o6) − cos (a x+ b)

a

5.11.2 Intégrales

(%i7) integrate(2*x+3, x, 0, 1);

(%o7) 4

(%i8) integrate(2*x^2+4*x-5,x, 1, 4);

(%o8) 57

(%i9) integrate(exp(-x+3),x, 3, 5);

(%o9) 1− e−2

(%i10) integrate(log(x-5),x, 6, 7);

(%o10) 2 log (2)− 1

(%i11) integrate(cos(2*x),x, 0, %pi/3);

(%o11)
√

3

4

(%i12) integrate(sin(2*x),x, 0, %pi/3);

(%o12)
3

4

5.12 Développements limités

On note N l’ordre des développements que l’on veut étudier.

(%i13) N:4;

(%o13) 4

La fonction exponentielle :

(%i14) taylor(exp(x), x, 0, N);

(%o14)/T/1 + x+
x2

2
+
x3

6
+
x4

24
+ ...

La fonction inverse :

(%i15) taylor(1/(1+x), x, 0, N);

(%o15)/T/1− x+ x2 − x3 + x4 + ...

La fonction logarithme népérien :

(%i16) taylor(log(1+x), x, 0, N);

(%o16)/T/x− x2

2
+
x3

3
− x4

4
+ ...

La fonction sinus :

(%i17) taylor(sin(x), x, 0, N);

(%o17)/T/x− x3

6
+ ...

La fonction cosinus :

(%i18) taylor(cos(x), x, 0, N);

(%o18)/T/1− x2

2
+
x4

24
+ ...

La fonction puissance :

(%i19) taylor((1+x)^alpha, x, 0, N);

(%o19)/T/1 + αx+

(
α2 − α

)
x2

2
+

(
α3 − 3α2 + 2α

)
x3

6
+

(
α4 − 6α3 + 11α2 − 6α

)
x4

24
+ ...

Chapitre 6

Les suites

6.1 Définition d’une suite

Suites définies explicitement, comme une fonction de n :

(%i1) a(n):=n^2;

(%o1) a (n) := n2

ou comme une liste infinie :

(%i2) a[n]:=n^2;

(%o2) an := n2

Suites définies par une relation de récurrence, comme une fonction de n :

(%i3) a(n):= if n = 0 then 4 else 2*a(n-1)-3$
a(110);

(%o4) 1298074214633706907132624082305027

ou comme une liste infinie :

(%i5) a[n]:= if n = 0 then 4 else 2*a[n-1]-3$
a[110];

(%o6) 1298074214633706907132624082305027

6.2 Expression explicite d’une suite définie par récurrence

On charge le module « solve_rec » :

(%i7) load("solve_rec");

(%o7) C : /PROGRA 2/MAXIMA 1.0−2/share/maxima/5.28.0−2/share/solve_rec/solve_rec.mac

(%i8) solve_rec(u[n]=2*u[n-1]-3, u[n]);

(%o8) un = %k1 2n − 3 2n + 3

Avec un terme initial :

37

(%i9) solve_rec(u[n]=2*u[n-1]-3, u[n], u[0]=4);

(%o9) un = 2n + 3

On peut définir une suite de cette façon :

(%i10) define(v[n], rhs(%o9));

(%o10) vn := 2n + 3

(%i11) v[9];

(%o11) 515

6.3 Représentation géométrique d’une suite

(%i12) entiers:makelist(n,n,0, 20);
termesdelasuite:map(a,entiers);
wxplot2d([discrete, entiers,termesdelasuite],[style, points]);

(%o12) [0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20]

(%o13) [4, 5, 7, 11, 19, 35, 67, 131, 259, 515, 1027, 2051, 4099, 8195, 16387, 32771, 65539, 131075, 262147, 524291, 1048579]

(%t14)

(%o14)

6.4 Suites arithmétiques

(%i15) r:1.23$
a(n):= if n < 1 then 0 else a(n-1)+r$
a(110);

(%o17) 135.3000000000001

(%i18) entiers:makelist(n,n,0, 20)$
termesdelasuite:map(a,entiers)$
wxplot2d([discrete, entiers,termesdelasuite],[style, points])$

(%t20)

6.5 Suites géométriques

(%i21) q:1.23$
a(n):= if n < 1 then 2 else a(n-1)*q$
a(10);

(%o23) 15.85189219210378

(%i24) entiers:makelist(n,n,0, 20)$
termesdelasuite:map(a,entiers)$
wxplot2d([discrete, entiers,termesdelasuite],[style, points])$

(%t26)

6.6 Limite d’une suite

(%i27) a(n):=n^2;
limit(a(n),n,inf);

(%o27) a (n) := n2

(%o28) ∞

(%i29) a(n):=1/n^2;
limit(a(n),n,inf);

(%o29) a (n) :=
1

n2

(%o30) 0

6.7 Somme d’une suite

On utilise la commande sum :

(%i31) sum(a(k), k, 1, 10);

(%o31)
1968329

1270080

6.8 Le module functs

(%i32) load("functs");

(%o32) C : /PROGRA 2/MAXIMA 1.0−2/share/maxima/5.28.0−2/share/simplification/functs.mac

Suites arithmétiques arithmetic(a, r, n) retourne le n-ième terme de la suite arithmétique de raison
r et de premier terme a :

(%i33) arithmetic(1,2,3);

(%o33) 5

geometric(a, q, n) retourne le n-ième terme de la suite arithmétique de raison r et de premier terme
a :

(%i34) geometric(1,2,3);

(%o34) 4

Somme des termes des suites arithmétiques et géométriques :

(%i35) arithsum(1,2,3);

(%o35) 9

(%i36) geosum(1,2,3);

(%o36) 7



Chapitre 7

Les nombres complexes

7.1 Forme algébrique d’un nombre complexe

Définition d’un nombre complexe

(%i1) z1:2+3*%i;

(%o1) 3 i+ 2

(%i2) z2:5+3*%i;

(%o2) 3 i+ 5

Somme et soustraction de deux nombres complexes

(%i3) z1+z2;
z1-z2;

(%o3) 6 i+ 7

(%o4) − 3

Produit de deux nombres complexes

Il faut développer l’expression avec expand si l’on souhaite avoir la forme algébrique :

(%i5) z1*z2;
expand(z1*z2);

(%o5) (3 i+ 2) (3 i+ 5)

(%o6) 21 i+ 1

Pour avoir la forme algébrique d’un nombre complexe, on utilise la fonction rectform :

(%i7) z1/z2;
rectform(z1/z2);

(%o7)
3 i+ 2

3 i+ 5

(%o8)
9 i

34
+

19

34

Une nouvelle identité

41

(%i9) declare(a,mainvar)$
expand((a-b*%i)*(a+b*%i));

(%o10) a2 + b2

Partie réelle

(%i11) realpart(z1);

(%o11) 2

Partie imaginaire

(%i12) imagpart(z1);

(%o12) 3

7.2 Représentation géométrique d’un nombre complexe

Définition d’une fonction utilisée pour placer les images des nombres complexes dans le plan com-
plexe :

(%i13) plotcomplex(Liste):=wxplot2d([discrete, makelist(realpart(Liste[k]), k, 1, length(Liste)),
makelist(imagpart(Liste[k]), k, 1, length(Liste))],[style, points])$

(%i14) plotcomplex([1+%i, 2+3*%i, -1-%i, 1.5+2*%i, 0]);

(%t14)

(%o14)

7.3 Nombre complexe conjugué

(%i15) conjugate(1+%i);

(%o15) 1− i

Représentation géométrique d’un conjugué

(%i16) liste:[1+%i, 2+3*%i, %i, -3+%i, 3*%i, -2]$
liste:append(liste, makelist(conjugate(liste[k]), k, 1, length(liste)))$
plotcomplex(liste);

(%t18)

(%o18)

(%i19) conjugate[liste];

(%o19) conjugate[i+1,3 i+2,i,i−3,3 i,−2,1−i,2−3 i,−i,−i−3,−3 i,−2]

7.4 Forme trigonométrique d’un nombre complexe

Définition d’un nombre complexe

(%i20) z:1+%i;

(%o20) i+ 1

Le module

(%i21) cabs(z);

(%o21)
√

2

Un argument

(%i22) carg(z);

(%o22)
π

4

Définition de la forme trigonométrique

(%i23) formetrigo(z):=[cabs(z), carg(z)]$

(%i24) formetrigo(z);

(%o24) [
√

2,
π

4
]

(%i25) formetrigo(1+%i*sqrt(3));

(%o25) [2,
π

3
]



44 CHAPITRE 7. LES NOMBRES COMPLEXES

7.5 Forme exponentielle

(%i26) polarform(z);

(%o26)
√

2 e
i π
4

(%i27) rectform(%);

(%o27) i+ 1

(%i28) z:exp(%i*theta);

(%o28) ei θ

Définition de eiθ

(%i29) rectform(z);

(%o29) i sin (θ) + cos (θ)

Cas particulier : eiπ + 1 = 0

(%i30) exp(%i*%pi);

(%o30) − 1

Illustration de l’égalité précédente et lim
n→+∞

(
1 +

z

n

)n
= ez

(%i31) N:64$
liste:makelist((1+%i*%pi/N)^k, k, 0, N)$
plotcomplex(liste);

(%t33)

(%o33)

Propriétés de la forme exponentielle

(%i34) exp(%i*theta[1])*exp(%i*theta[2]);

(%o34) ei θ2+i θ1

(%i35) 1/exp(%i*theta);

(%o35) e−i θ

(%i36) exp(%i*theta[1])/exp(%i*theta[2]);

(%o36) ei θ1−i θ2

(%i37) exp(%i*theta)^n;

(%o37) ei n θ

Formule de Moivre : (i sin(θ) + cos(θ))n = i sin(nθ) + cos(nθ)

(%i38) (cos(theta)+%i*sin(theta))^n;

(%o38) (i sin (θ) + cos (θ))
n

(%i39) trigrat((exp(%i*theta))^n);

(%o39) i sin (n θ) + cos (n θ)

En une seule commande

(%i40) trigrat((cos(theta)+%i*sin(theta))^n);

(%o40) i sin (n θ) + cos (n θ)

Formules d’Euler

(%i41) exponentialize(cos(theta));

(%o41)
ei θ + e−i θ

2

(%i42) exponentialize(sin(theta));

(%o42) −
i
(
ei θ − e−i θ

)
2

7.6 Équations du second degré à coefficients constants

Maxima ne fait pas la distinction entre delta positif ou négatif :

(%i43) kill(z)$

(%i44) solutions:solve(a*z^2+b*z+c=0, z);

(%o44) [z = −
√
b2 − 4 c a+ b

2 a
, z =

√
b2 − 4 c a− b

2 a
]

(%i45) solve(z^2+z-1=0,z);

(%o45) [z = −
√

5 + 1

2
, z =

√
5− 1

2
]

(%i46) solve(z^2+2*z+1=0,z);

(%o46) [z = −1]

(%i47) solve(z^2+z+1=0,z);

(%o47) [z = −
√

3 i+ 1

2
, z =

√
3 i− 1

2
]

Pour vérifier qu’un nombre complexe annule une expression :

(%i48) subst(-1, z, z^2+z+1);

(%o48) 1



Chapitre 8

Listes et statistiques descriptives

8.1 Listes

Définition d’une liste :

(%i1) L:[30, 14, 15, 16, 14, 15, 18, 19, 1 , 6 ,18 ];

(%o1) [30, 14, 15, 16, 14, 15, 18, 19, 1, 6, 18]

Longueur d’une liste :

(%i2) length(L);

(%o2) 11

Concaténer deux listes :

(%i3) append(L,[0, 2, 3]);

(%o3) [30, 14, 15, 16, 14, 15, 18, 19, 1, 6, 18, 0, 2, 3]

Ordonner la liste :

(%i4) sort(L);

(%o4) [1, 6, 14, 14, 15, 15, 16, 18, 18, 19, 30]

Créer un tableau de valeur d’une fonction :

(%i5) f(x):=x^2+1;

(%o5) f (x) := x2 + 1

On crée une liste de nombres régulièrement espacés, par exemple de 0,1 :

(%i6) nombres:makelist(k/10,k,0,10);

(%o6) [0,
1

10
,

1

5
,

3

10
,

2

5
,

1

2
,

3

5
,

7

10
,

4

5
,

9

10
, 1]

On crée la liste des images par f de ces nombres :

(%i7) fnombres:f(nombres);

(%o7) [1,
101

100
,

26

25
,

109

100
,

29

25
,

5

4
,

34

25
,

149

100
,

41

25
,

181

100
, 2]

47

Si on souhaite avoir les valeurs approchées :

(%i8) anombres:float(fnombres);

(%o8) [1.0, 1.01, 1.04, 1.09, 1.16, 1.25, 1.36, 1.49, 1.64, 1.81, 2.0]

8.2 Statistiques descriptives

(%i9) load(descriptive);

(%o9) C : /PROGRA 2/MAXIMA 1.0−2/share/maxima/5.28.0−2/share/descriptive/descriptive.mac

Valeur moyenne de L :

(%i10) mean(L);

(%o10)
166

11

(%i11) float(mean(L));

(%o11) 15.09090909090909

Écart-type :

(%i12) std(L);

(%o12)
2
√

1482

11

(%i13) float(std(L));

(%o13) 6.999409656334603

Minimum de L :

(%i14) mini(L);

(%o14) 1

Macimum de L :

(%i15) maxi(L);

(%o15) 30

Étendue :

(%i16) range(L);

(%o16) 29

Médiane :

(%i17) median(L);

(%o17) 15

Quantiles :

(%i18) quantile(L,1/4);

(%o18) 14

(%i19) quantile(L,2/4);

(%o19) 15

(%i20) quantile(L,3/4);

(%o20) 18

(%i21) quantile(L,1/10);

(%o21) 6

8.3 Listes avec pondérations

Liste des valeurs :

(%i22) L:[12,13,15,16];

(%o22) [12, 13, 15, 16]

Liste des effectifs ou des fréquences :

(%i23) F:[1,2,3,4];

(%o23) [1, 2, 3, 4]

(%i24) moyenne(L_1, L_2):=block ( [n:length(L_1)],

sum(L_1[k]*L_2[k],k,1,n)/sum(L_2[k],k,1,n)
)$

(%i25) variance(L_1, L_2):=block ( [n:length(L_1)],

moyenne(L_1^2,L_2)-moyenne(L_1, L_2)^2
)$

(%i26) ecartype(L_1, L_2):=sqrt(variance(L_1, L_2))$

(%i27) moyenne(L,F);
variance(L,F);
ecartype(L,F);

(%o27)
147

10

(%o28)
201

100

(%o29)
√

201

10

Chapitre 9

Probabilités

9.1 Simulations

9.1.1 Simulation du lancer de pièce et loi de Bernoulli

On utilise la commande random(2), pour avoir 0 ou 1 de façon aléatoire et équiprobable :

(%i1) random(2);
random(2);
random(2);

(%o1) 0

(%o2) 0

(%o3) 0

Liste de simulations du lancer de pièce

(%i4) Liste:[]$
N:30$
for i from 1 thru N do Liste:append(Liste,[random(2)])$
Liste;
entiers:makelist(n,n,1,N)$
wxplot2d([discrete, entiers, Liste], [style, points]);

(%o7) [1, 0, 1, 1, 1, 0, 1, 1, 1, 0, 0, 1, 0, 1, 0, 1, 0, 1, 1, 1, 1, 0, 0, 0, 1, 0, 1, 0, 1, 1]

(%t9)

(%o9)

51

Affichage des fréquences

(%i10) zeroUn:[0, 1]$
f1:sum(Liste[k],k, 1, N)/N$
f0:1-f1$
float(f0);
float(f1);
ListeFrequences:[f0, f1]$
wxplot2d([discrete, zeroUn, ListeFrequences], [style, points]);

(%o13) 0.4

(%o14) 0.6

(%t16)

(%o16)

La commande random(1.0) permet d’obtenir un nombre de l’intervalle [0; 1[.

Pour obtenir un lancer de pièce truquée, où la probabilité d’obtenir le 1 est p, on utilise la commande
ci-dessous :

(%i17) p:0.3$
random(1.0)+ p;

(%o18) 1.152705109085575

(%i19) Liste:[]$
N:30$
for i from 1 thru N do Liste:append(Liste,[floor(random(1.0)+p)])$
Liste;
entiers:makelist(n,n,1,N)$
wxplot2d([discrete, entiers, Liste], [style, points]);

(%o22) [1, 0, 1, 0, 0, 0, 0, 1, 1, 0, 1, 0, 1, 0, 0, 0, 1, 1, 0, 1, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0]

(%t24)

(%o24)

Affichage des fréquences

(%i25) zeroUn:[0, 1]$
f1:sum(Liste[k],k, 1, N)/N$
f0:1-f1$
float(f0);
float(f1);
ListeFrequences:[f0, f1]$
wxplot2d([discrete, zeroUn, ListeFrequences], [style, points]);

(%o28) 0.66666666666667

(%o29) 0.33333333333333

(%t31)

(%o31)

9.1.2 Simulation du lancer de dé

Simulation du lancer de dé

(%i32) random(7);
random(7);
random(7);
random(7);
random(7);
random(7);
random(7);

(%o32) 0

(%o33) 3

(%o34) 5

(%o35) 1

(%o36) 0

(%o37) 4

(%o38) 3

Liste de simulation du lancer de dé

(%i39) Liste:[]$
N:1000$
for i from 1 thru N do Liste:append(Liste,[random(6)+1])$
Liste$
entiers:makelist(n,n,1,N)$
wxplot2d([discrete, entiers, Liste], [style, points], [y,0,6])$

(%t44)

Transformation d’une liste en un ensemble

(%i45) set:setify(Liste);
for s in set do print (s);

(%o45) 1, 2, 3, 4, 5, 6123456

(%o46) done

Calcul des fréquences

(%i47) listeFrequences:[]$
for element in set do (

s:0,
for i step 1 from 1 thru length(Liste) do (

if element=Liste[i] then s:s+1
),
listeFrequences:append(listeFrequences, [s/length(Liste)])

)$
float(listeFrequences);
entiers:makelist(n,n,1,6)$
wxplot2d([discrete, entiers, listeFrequences], [style, points], [y,0,1]);

(%o49) [0.185, 0.159, 0.157, 0.165, 0.162, 0.172]

(%t51)

(%o51)

9.1.3 Simulation de la loi binomiale

(%i52) nombreSucces(N,p):=block([],
somme:0,
for i from 1 thru N do (

somme:somme+floor(random(1.0)+p)
),

somme
)$

(%i53) nombreSucces(30,0.7);

(%o53) 26

(%i54) Liste:[]$
N:300$
p:0.7$
for i from 1 thru N do Liste:append(Liste,[nombreSucces(N,p)])$
entiers:makelist(n,n,1,N)$
wxplot2d([discrete, entiers, Liste], [style, points]);

(%t59)

(%o59)

(%i60) listeFrequences:[]$
listeValeurs:makelist(i,i,0,N)$
for element in setify(listeValeurs) do (

s:0,
for i step 1 from 1 thru length(Liste) do (

if element=Liste[i] then s:s+1
),
listeFrequences:append(listeFrequences, [s/length(Liste)])

)$
wxplot2d([discrete, listeValeurs, listeFrequences], [style, points],[y,0,0.1]);

(%t63)

(%o63)

Définition de la fonction simulation de la loi binomiale, elle retourne une liste :

(%i64) declare(N, integer)$
assume(N>0)$
simulationLoiBinomiale(N,p):=block([],

Liste:[],
for i from 1 thru N do Liste:append(Liste,[nombreSucces(N,p)]),
listeFrequences:[],
listeValeurs:makelist(i,i,0,N),
for element in setify(listeValeurs) do (

s:0,
for i step 1 from 1 thru length(Liste) do (

if element=Liste[i] then s:s+1
),

listeFrequences:append(listeFrequences, [s/length(Liste)])
),

listeFrequences
)$

Définition de la fonction affichage d’une simulation de la loi binomiale :

(%i67) afficheSimulationLoiBinomiale(N,p):=block([],
listeFrequences:simulationLoiBinomiale(N,p),
wxplot2d([discrete, makelist(i,i,0,N), listeFrequences], [style, points])

)$

(%i68) afficheSimulationLoiBinomiale(40,0.8);

(%t68)

(%o68)

9.1.4 Simulation de la loi uniforme

Tirage aléatoire d’un nombre de l’intervalle [0; 1[

(%i69) random(1.0);

(%o69) 0.95659153069339

Création d’une liste de nombres aléatoires de l’intervalle [0; 1[

(%i70) N:100$
Liste:[]$
for i from 0 thru N do Liste:append(Liste,[random(1.0)])$
Liste$

Affichage de la liste précédente

(%i74) entiers:makelist(n,n,0,N)$
wxplot2d([discrete, entiers, Liste], [style, points],[y, 0, 1]);

(%t75)

(%o75)

9.2 Distributions

(%i76) load(distrib)$

9.2.1 Schéma de Bernoulli

À retenir :

pdf : probability density function, c’est la fonction densité de probabilité

cdf : cumulative density function, c’est la fonction de répartition pdf_bernoulli(1,p) donne la valeur
de P (X = 1) où X suit la loi de Bernoulli de paramètre p :

(%i77) assume(0<p,p<1)$
pdf_bernoulli(1,p);
pdf_bernoulli(0,p);

(%o78) 0.7

(%o79) 0.3

Représentation graphique

(%i80) p:0.3;
zeroUn:[0, 1]$
f1:p$
f0:1-f1$
float(f0);
float(f1);
ListeFrequences:[f0, f1]$
wxplot2d([discrete, zeroUn, ListeFrequences], [style, points]);

(%o80) 0.3

(%o84) 0.7

(%o85) 0.3

(%t87)

(%o87)

9.2.2 Loi binomiale

pdf_binomial(k,n,p) ; donne la valeur de P (X = k) où X suit la loi binomiale de paramètres n et
p :

(%i88) pdf_binomial(5,7,1/6);

(%o88)
175

93312

cdf_binomial(k,n,p) ; donne la valeur de P (X ≤ k) où X suit la loi binomiale de paramètres n et
p :

(%i89) cdf_binomial(5,7,1/6);

(%o89)
7775

7776

Calcul de P (a ≤ X ≤ b)

(%i90) n:20$
p:0.3$
a:2$
b:5$
cdf_binomial(b,n,p)-cdf_binomial(a-1,n,p);

(%o94) 0.40873356967328

Coefficient binomial

(%i95) binomial (5, 2);

(%o95) 10

Représentation graphique de la loi binomiale

(%i96) N:30;
entiers:makelist(n,n,0,N)$
listeBinomiale:makelist(pdf_binomial(n,N,0.7),n,0,N)$
wxplot2d([discrete, entiers, listeBinomiale], [style, points], [y,0,0.5]);

(%o96) 30

(%t99)

(%o99)

Comparaison avec une simulation

(%i100)N:30;
entiers:makelist(n,n,0,N)$
listeBinomiale:makelist(pdf_binomial(n,N,0.7),n,0,N)$
wxplot2d([[discrete, entiers, listeBinomiale], [discrete, entiers, simulationLoiBinomiale(N,0.7)]], [style, points], [y,0,0.5]);

(%o100)30

(%t103)

(%o103)

Représentation graphique de la fonction de répartition de loi binomiale

(%i104)N:30;
entiers:makelist(n,n,0,N)$
listeBinomiale:makelist(cdf_binomial(n,N,0.7),n,0,N)$
wxplot2d([discrete, entiers, listeBinomiale], [style, points], [y,0,1]);

(%o104)30

(%t107)

(%o107)

9.2.3 Loi de Poisson

float(pdf_poisson(k,lambda)) ; donne une valeur approchée de P (X = k) = λk

k! e
( − λ) :

(%i108)float(pdf_poisson(3,95/100));

(%o108)0.055263680830717

float(cdf_poisson(k,lambda)) ; donne une valeur approchée de P (X ≤ k) :

(%i109)float(cdf_poisson(3,95/100));

(%o109)0.98392556340084

Calcul de P (a ≤ X ≤ b)

(%i110)lambda:7.3$
a:2$
b:5$
cdf_poisson(b,lambda)-cdf_poisson(a-1,lambda);

(%o113)0.258436232947

Représentation graphique de la loi de Poisson

(%i114)N:30;
entiers:makelist(n,n,0,N)$
listePoisson:makelist(pdf_poisson(n,21),n,0,N)$
wxplot2d([discrete, entiers, listePoisson], [style, points], [y,0,0.5]);

(%o114)30

(%t117)

(%o117)

Représentation graphique de la fonction de répartition de loi de Poisson

(%i118)N:30;
entiers:makelist(n,n,0,N)$
listePoisson:makelist(cdf_poisson(n,21),n,0,N)$
wxplot2d([discrete, entiers, listePoisson], [style, points], [y,0,1]);

(%o118)30

(%t121)

(%o121)

9.2.4 Loi normale

float(pdf_normal(k,m,s)) ; donne une valeur approchée de la fonction densité de la loi normale
d’espérance m et d’écart-type s :

(%i122)float(pdf_normal(95/100,0,1));

(%o122)0.25405905646919

float(cdf_normal(k,m,s)) ; donne une valeur approchée de la fonction répartition de la loi normale
d’espérance m et d’écart-type s :

(%i123)float(cdf_normal(95/100,0,1));

(%o123)0.82894387369152

Calcul de P (a ≤ X ≤ b)

(%i124)m:0.95$
s:1.5$
a:2$
b:5$
float(cdf_normal(b,m,s)-cdf_normal(a,m,s));

(%o128)0.23849667842003

Représentation graphique de la loi normale Loi normale centrée réduite

(%i129)wxplot2d(pdf_normal(x,0,1),[x,-5,5], [y,0,0.6]);

(%t129)

(%o129)

Loi normale de paramètres 21 et 6.3 (variance)

(%i130)wxplot2d(pdf_normal(x,21,sqrt(6.3)),[x,0,30], [y,0,0.5]);

(%t130)

(%o130)

Représentation graphique de la fonction de répartition de la loi normale Fonction de répartition de
la loi normale centrée réduite

(%i131)wxplot2d(cdf_normal(x,0,1),[x,-5,5], [y,0,1]);

(%t131)

(%o131)

F

(%i132)wxplot2d(cdf_normal(x,21,sqrt(6.3)),[x,0,30], [y,0,1]);

(%t132)

(%o132)

9.2.5 Approximation d’une loi binomiale par une loi normale

(%i133)N:30;
p:0.7;
N*p;
N*(1-p);
entiers:makelist(n,n,0,N)$
listeBinomiale:makelist(pdf_binomial(n,N,p),n,0,N)$
wxplot2d([[discrete, entiers, listeBinomiale], pdf_normal(x,N*p,sqrt(N*p*(1-p)))],

[x,0,N],
[style, points, lines],
[y,0,0.2],
[legend, "Loi binomiale", "Loi normale"],
[xlabel, "x"],

[ylabel, "y"]
);

(%o133)30

(%o134)0.7

(%o135)21.0

(%o136)9.000000000000002

(%t139)

(%o139)

9.2.6 Approximation d’une loi binomiale par une loi de Poisson

(%i140)N:40;
p:0.09;
N*p*(1-p);
entiers:makelist(n,n,0,N)$
listeBinomiale:makelist(pdf_binomial(n,N,p),n,0,N)$
listePoisson:makelist(pdf_poisson(n,N*p),n,0,N)$
wxplot2d([[discrete, entiers, listeBinomiale], [discrete, entiers, listePoisson]],

[x,0,N],
[style, points, points],
[y,0,0.25],
[legend, "Loi binomiale", "Loi de Poisson"],
[xlabel, "x"],

[ylabel, "y"]
);

(%o140)40

(%o141)0.09

(%o142)3.276

(%t146)

(%o146)

Chapitre 10

Algorithmique et programmation

10.1 Boucle Tant que

Boucle << tant que >> (<< while >>)

(%i1) n:0$
s:0$
while n < 10 do (

n : n + 1,
s : s + n

)$
s;

(%o4) 55

10.2 Boucle Pour

Boucle << pour >> (<< for >>), le pas est égal à 1 par défaut :

(%i5) s:0$
for n from 0 thru 10 do (

s : s + n
)$
s;

(%o7) 55

On peut écrire la ligne for en une seule ligne :

(%i8) s:0$
for n from 0 thru 10 do s : s + n$
s;

(%o10) 55

Avec un pas de 2

(%i11) s:0$
for n step 2 from 0 thru 10 do s : s + n$
s;

(%o13) 30

10.3 Test

Test : si, alors, sinon (if, then, else)

(%i14) f(x):=if x < 0 then 0 else 1;

(%o14) f (x) := ifx < 0then0else1

(%i15) wxplot2d(f(x),[x, -4,4], [y, -1, 2]);

(%t15)

(%o15)

Sans le else :

(%i16) f(x):=if x < 0 then 0;
wxplot2d(f(x),[x, -4,4], [y, -1, 2]);

(%o16) f (x) := ifx < 0then0plot2d : expressionevaluatestonon−numericvaluesomewhereinplottingrange.

(%t17)

(%o17)

Récursivité définition de la factorielle (pour x entier)

(%i18) f(x):= if x < 1 then 1 else x*f(x-1);

(%o18) f (x) := ifx < 1then1elsex f (x− 1)

(%i19) wxplot2d(f(x),[x, -2,4], [y, -1, 20]);

plot2d : somevalueswereclipped.

(%t19)

(%o19)

10.4 Logique (utile pour les tests)

Utilisation de vrai et faux

(%i20) true and flase;

(%o20) flase

(%i21) true or flase;

(%o21) true

Avec des parenthèses : associativité

(%i22) (true and flase) or true;

(%o22) true

Avec des variables

(%i23) x:true$
y:false$
x and y;
x or y;
kill(x, y);

(%o25) false

(%o26) true

(%o27) done

10.5 Structure d’une fonction/procédure

Définition d’une fonction (une valeur est retournée par un return)

nom(paramètres d’entrées):=block([paramètres locaux],
block d’instructions 1,
/* Commentaires */
block d’instructions 2,
return(valeur)

);

Exemple : définition de la factorielle (pour x entier).

(%i28) g(x) := block([temp],
temp:1,
while x > 1 do(

temp:x*temp,
x:x-1),

return(temp))$

(%i29) g(5);

(%o29) 120

Définition d’une procédure (la dernière valeur est retournée)

nom(paramètres d’entrées):=block([paramètres locaux],
block d’instructions 1,
/* Commentaires */
block d’instructions 2

);

Exemple : définition de la factorielle (pour x entier).

(%i30) g(x) := block([temp],
temp:1,
while x > 1 do(

temp:x*temp,
x:x-1),

temp)$

(%i31) g(5);

(%o31) 120

Un procédure peut être utilisée si l’on souhaite par exemple afficher une courbe et que l’on a rien
à retourner.

(%i32) courbeExponentielle(n) := block([],
wxplot2d(exp(n*x), [x, -2, 2])

)$

(%i33) courbeExponentielle(2);

(%t33)

(%o33)

(%i34) courbeExponentielle(-0.5);

(%t34)

(%o34)

10.6 Affichage de chaînes de caractères.

(%i35) test(x,y):=block([],
if x > y then print(x, "est plus grand que ", y)

else print(x, "n’est pas plus grand que", y),
return("Commande exécutée"))$

(%i36) test(4,3);

4estplusgrandque3

(%o36) Commandeexcute

Il y a aussi la fonction display qui permet d’afficher le contenu d’une variable.

(%i37) x:4;
display(x);

(%o37) 4x = 4

(%o38) done

10.7 Utilisations d’hypothèses

Pour oublier les hypothèses ou les affectations de y :

(%i39) kill(y);

(%o39) done

(%i40) assume(y > 0);

(%o40) [y > 0]

(%i41) test(y^2+1,-y);

y2 + 1estplusgrandque− y

(%o41) Commandeexcute

Pour oublier l’hypothèse :

(%i42) forget(y>0);

(%o42) [y > 0]

(%i43) test(y^2+1,-y);

(%o43) Commandeexcute

(%i44) assume(y < 0);
test(y^2+1,-y);

(%o44) [y < 0]

(%o45) Commandeexcute

(%i46) facts();

(%o46) [0 > y]

(%i47) forget(all);

(%o47) [true]

(%i48) facts();

(%o48) [0 > y]

10.8 Types de variables

Si n est un entier, on peut écrire :

(%i49) declare(n, integer);
assume(n>0);

(%o49) done

(%o50) [redundant]

Cela peut changer des calculs :

(%i51) cos(n*%pi);

(%o51) 1

Si on oublie tout :

(%i52) kill(all);

(%o0) done

(%i1) cos(n*%pi);

(%o1) cos (π n)

10.9 Commentaires

(%i2) /* Ceci est un commentaire,
la somme de deux entiers ! */

1+2;

(%o2) 3

Chapitre 11

Équations différentielles

11.1 Équations différentielles du premier ordre

On définit une équation différentielle et son équation homogène :

(%i1) equadiff:’diff(y,x)+y=exp(x);
equadiff0:lhs(equadiff)=0;

(%o1)
d

d x
y + y = ex

(%o2)
d

d x
y + y = 0

Menu/Équations/Résoudre une équation différentielle...

(%i3) ode2(equadiff, y, x);
ode2(equadiff0, y, x);

(%o3) y = e−x
(
e2 x

2
+ %c

)
(%o4) y = %c e−x

où la fonction inconnue est notée y, sa dérivée ’diff(y,x) et %c désigne un réel quelconque. On peut
développer le résultat précédent en utilisant % et expand (% permet de récupérer le dernier résultat
calculé) :

(%i5) expand(%);

(%o5) y = %c e−x

Si par exemple on a la condition initiale y(0)=1 : Menu/Équations/Condition initiele (1)...

(%i6) ic1(%, x=0, y=1);

(%o6) y = e−x

On peut développer :

(%i7) expand(%);

(%o7) y = e−x

75



76 CHAPITRE 11. ÉQUATIONS DIFFÉRENTIELLES

On peut vérifier que l’on obtient bien une solution de l’équation différentielle. La commande sui-
vante remplace y par son expression dans l’équation différentielle, diff permet de calculer les déri-
vées.

(%i8) ev(equadiff, %o6, diff);

(%o8) 0 = ex

On a l’égalité donc c’est bien une solution de l’équation différentielle.

11.2 Équations différentielles du second ordre

Menu/Équations/Résoudre une équation différentielle...

(%i9) ode2(’diff(y,x,2)+4*’diff(y,x)+4*y=1-exp(-x), y, x);

(%o9) y =
e−x (ex − 4)

4
+ (%k2x+ %k1) e−2 x

où la fonction inconnue est notée y, sa dérivée ’diff(y,x), sa dérivée seconde ’diff(y,x,2) et %k1 et
%k2 désignent deux réels quelconques. On peut développer

(%i10) expand(%o9);

(%o10) y = −e−x + %k2x e−2 x + %k1 e−2 x +
1

4

Si on a les conditions initiales y(0)=1 et y’(0)=2 : Menu/Équations/Condition initiele (2)...

(%i11) ic2(%o9, x=0, y=1, ’diff(y,x)=2);

(%o11) y =
e−x (ex − 4)

4
+

(
9x

2
+

7

4

)
e−2 x

Si on a les conditions initiales y(0)=1 et y(1)=3 : Menu/Équations/Problème aux limites...

(%i12) bc2(%o9,x = 0,y = 1,x = 1,y = 3);

(%o12) y =
e−x (ex − 4)

4
+

((
11 e2 + 4 e− 7

)
x

4
+

7

4

)
e−2 x

11.3 Exemple : masse - ressort - frottement visqueux

(%i13) equation:m*’diff(y,x,2)+c*’diff(y,x)+r*y=1;

(%o13) m
(
d2

d x2
y

)
+ c

(
d

d x
y

)
+ r y = 1

Pour delta positif où m est la masse, c est le coefficient de frottement, r la raideur, et le second
membre de l’équation différentielle est donné par une force externe appliquée à la masse

(%i14) forget(facts());
assume(c**2-4*m*r>0);
ode2(equation, y, x);

(%o14) [[]]

(%o15) [c2 > 4mr]

(%o16) y = %k1 e

(√
c2

m2 − 4 r
m

− c
m

)
x

2 + %k2 e

(
−
√
c2

m2 − 4 r
m

− c
m

)
x

2 +
1

r

Pour delta nul

(%i17) forget(facts());
assume(equal(c**2-4*m*r,0));
ode2(equation, y, x);

(%o17) [[c2 > 4mr]]

(%o18) [equal
(
c2, 4mr

)
]

(%o19) y = (%k2x+ %k1) e−
c x
2m +

4m

c2

Pour delta négatif

(%i20) forget(facts());
assume(c**2-4*m*r<0);
ode2(equation, y, x);

(%o20) [[equal
(
c2, 4mr

)
]]

(%o21) [4mr > c2]

(%o22) y = e−
c x
2m

%k1 sin


√

4 r
m −

c2

m2 x

2

+ %k2 cos


√

4 r
m −

c2

m2 x

2

+
1

r

Chapitre 12

Séries de Fourier

Pour le paragraphe « Utilisation du package piecewise », il faut utiliser une bibliothèque, pour la
récupérer, cliquer deux fois sur le lien ci-contre : lien .

Enregistrez ce fichier dans le répertoire de votre choix et n’oubliez pas de donner le chemin complet
lors de l’utilisation :

load("C:/chemin-complet/pw.mac");

12.1 Définition de la fonction périodique

On suppose que l’on se donne une fonction périodique de période T , définie sur R et qu’une seule
expression est donnée sur l’intervalle [a; b] de longueur T . Pour les fonctions définies par morceaux
sur l’intervalle [a; b], le lecteur est renvoyé à la section sur le package piecewise. La notation g
désignera la fonction définie sur [a; b]. La notation f désignera la fonction définie sur R.

(%i1) a:-%pi;
b:%pi;
T:b-a;
%omega:2*%pi/T;
g(t):=t;
f(t):=g(t-(b-a)*floor((t-a)/(b-a)));

(%o1) − π

(%o2) π

(%o3) 2π

(%o4) 1

(%o5) g (t) := t

(%o6) f (t) := g

(
t− (b− a) floor

(
t− a
b− a

))

/* Copyright (C) 2008-2011 by Richard V. Hennessy, rich.hennessy@verizon.net
 *
 * This file ("this code" or "software") is released under the terms of the GNU General Public License,
 * version 2.
 * The user of this code assumes all risk for its use. This software has NO WARRANTY,
 * not even the implied warranty of MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.
 * This software is registered and copyrighted with www.copyright.gov.

-----------------------------------------------------------------------------------------------------------
 */

put('pw, 6.5, 'version);
put('pw, "Richard Hennessy", 'author);
put('pw, "rtest_pw.mac", 'test_suite);
put('pw, "Copyright - Richard Hennessy, 2008-2012", 'copyright);

pwinfo([_package]):=
block(
	disp("package name: pw.mac (c)"),
	disp("author: Richard V. Hennessy"),
	disp("revision: 6.5"),
	disp("Test suite : rtest_pw.mac"),
	disp("Recommended location: share/contrib"),
	disp("last update: June, 2012"),
	disp("")
)$

(
	load("opsubst"),
	load("unwind_protect"),
	load("simplifying.lisp"),
	load("abs_integrate.mac")
);


simp_given(__e, [__fcts]) ::=
buildq
  (
	[__e,__fcts], 
	block
	(
		[_fct, _otherfacts:facts()],
		unwind_protect
		(
			(
				apply(assume, __fcts), 
				expand(__e, 0, 0)
			), 
			for _fct in facts() do 
			(
				if not member(_fct, _otherfacts) then
					apply('forget, ['_fct])
			)		
		)
	)
);

_usepwdelta:false;
global:true$
use_between:false;

/* default is not to generate delta's in the output from diff */
gradef(unit_step(__x), 0)$
gradef(signum(__x), 0)$
gradef(unit_spike(__x), 0)$
gradef(abs(x), signum(x))$
gradef(between(__x,__a,__b,__c), 0, und, und, und)$
gradef(unit_pulse(__x), 0)$

gen_deltas([_v]):=block
(
	if listp(_v) and not emptyp(_v) and is(first(_v) = false) then
	(
		_usepwdelta:false,
		gradef(unit_step(__x), 0),
		gradef(signum(__x), 0),
		gradef(unit_pulse(__x), 0),
		gradef(between(__x,__a,__b,__c), 0, und, und)
	)
	elseif listp(_v) and not emptyp(_v) and is(first(_v) = true) then
	(
		_usepwdelta:true,
		gradef(unit_step(__x), pwdelta(__x)),
		gradef(signum(__x), 2*pwdelta(__x)),
		gradef(unit_pulse(__x), pwdelta(__x)-pwdelta(__x-1)),
		gradef(between(__x,__a,__b,__c), (1-unit_spike(__b-__a)+signum(__b-__a))/2*(pwdelta(__x-__a)-pwdelta(__x-__b)), und, und, und)
	),
	if _usepwdelta then
		"pwdeltas() will be used"
	else
		"no pwdeltas() will be used"
)$

_useabsint:true;

_eim:copy(extra_integration_methods);
_edim:copy(extra_definite_integration_methods);

use_absint([_v]):=block
(
	if listp(_v) and not emptyp(_v) and is(first(_v) = false) then
	(
		extra_integration_methods:[],
		extra_definite_integration_methods:[],
		_useabsint:false
	)
	elseif listp(_v) and not emptyp(_v) and is(first(_v) = true) then
	(
		extra_integration_methods:copy(_eim),
		extra_definite_integration_methods:copy(_edim),
		_useabsint:true
	),
	if _useabsint then
		"additional integration methods from abs_integrate.mac will be used"
	else
		"additional integration methods from abs_integrate.mac will not be used"
)$


gradef(pwdelta(__x), diff_pwdelta(1,__x))$
gradef(diff_pwdelta(__nn,__x), und, diff_pwdelta(__nn+1,__x))$

inpargs(__e) := block([inflag : true], if mapatom(__e) then [__e] else substinpart("[",__e,0))$
freeofall(__vlist,__e) := block([_ans], _ans:true, for _v in __vlist while _ans do if not freeof(_v, __e) then _ans : false, _ans)$

/* iif stuff, be careful about this.  This is new code and it may have bugs in it.  */
not_op(__e):= psubst(['notequal='equal, 'equal='notequal, ">"="<=", ">="="<", "<"=">=", "<="=">"],__e);

simp_iif(__e) :=
block(
[_done:false,inflag:true],
	while not _done do (
		if is(equal(isimp(__e),__e)) # true then
			__e : isimp(__e)
		else
		_done:true),
__e);

isimp(__e):=
block(
	[opsubst:true, inflag:true],
	simp_given(subst(iif =
	lambda([__cnd,__a,__b],
	if is(equal(__a,__b))=true then
		__a
    elseif safe_op(__cnd) = 'equal then
	(
		iif(__cnd, ratsubst(inpart(__cnd, 2), inpart(__cnd, 1), __a), __b)
	)
    elseif safe_op(__cnd) = 'notequal then
	(
		iif(__cnd, __a, ratsubst(inpart(__cnd, 2), inpart(__cnd, 1), __b))
	)
	else
		iif(__cnd, __a, __b)
	), __e))
);

simpiif(__cnd, __a, __b):=
block(
	[prederror:false],
	if not member(safe_op(__cnd), [">", "<", ">=", "<=", 'equal, 'notequal]) then
		error("The only operators allowed are >, <, >=, <=, equal() and notequal().")
	elseif is(__cnd)=true then
		__a
	elseif is(__cnd)=false then
		__b
	else
		simpfuncall('iif, __cnd, simp_given(__a, __cnd), simp_given(__b, not_op(__cnd)))
)$

simpunitspike(__x):=
block(
	[prederror:false],
	if is(equal(__x,0))=true then
		1
	elseif is(notequal(__x,0))=true then
		0
	else
		simpfuncall('unit_spike,__x)
)$


simpunitpulse(__x):=
block(
	[prederror:false],
	if is(__x>0 and __x<1)=true then
		1
	elseif is(equal(__x,0))=true then
		1/2
	elseif is(equal(__x,1))=true then
		1/2
	elseif is(__x<0 or __x>1)=true then
		0
	else
		simpfuncall('unit_pulse,__x)
)$

simppwdelta([__e]):=
block(
	[prederror:false],
	if length(__e) # 1 then
		error("Error : pwdelta takes one argument.")
	elseif numberp(first(__e)) and is(notequal(first(__e),0))=true then
		0
	elseif numberp(first(__e)) and is(equal(first(__e),0))=true then
		'inf
	else
		simpfuncall('pwdelta, first(__e))
)$

simpdiffpwdelta([__e]):=
block(
	[prederror:false],
	if length(__e) # 2 then
		error("Error : diff_pwdelta takes two arguments.")
	elseif numberp(second(__e)) and is(notequal(second(__e),0))=true then
		0
	elseif numberp(second(__e)) and is(equal(second(__e),0))=true then
		'und
	else
		simpfuncall('diff_pwdelta, first(__e), second(__e))
)$

simpbetween(__x,__a,__b, [__option]):=
block(
	[prederror:false,compxa:false,compxb:false],
	__x:rationalize(__x),
	__a:rationalize(__a),
	__b:rationalize(__b),
	if length(__option) > 1 then
		error(concat("No version of between() takes ", length(__option)+3, " arguments"))
	elseif is(__a >= __b)=true then
		0
	elseif emptyp(__option) or first(__option) = 'halfopen then
		if (compxa:sign(__x - __a)) = 'pos and (compxb:sign(__x - __b)) = 'neg  then
			1
		elseif compxa = 'neg or compxb = 'pos then
			0
		elseif compxa = 'zero or compxb = 'zero then
			1/2
		elseif __b = 'inf then
		(
			(1 + signum(__x - __a))/2
		)
		elseif __a = 'minf then
		(
			(1 - signum(__x - __b))/2
		)
		else
			(simpfuncall('between, __x, __a, __b, 'halfopen))
	elseif first(__option) = 'closed then
		if member(compxa:sign(__x - __a), '[pos,pz,zero]) and member(compxb:sign(__x - __b),'[neg,nz,zero]) then
			1
		elseif compxa='neg or compxb='pos then
			0
		elseif __b = 'inf then
		(
			(1 + signum(__x - __a) + unit_spike(__x - __a))/2
		)
		elseif __a = 'minf then
		(
			(1 - signum(__x - __b) + unit_spike(__x - __b))/2
		)
		else
			(simpfuncall('between, __x, __a, __b, first(__option)))
	elseif first(__option) = 'open then
		if (compxa:sign(__x - __a)) = 'pos and (compxb:sign(__x - __b)) = 'neg then
			1
		elseif member(compxa, '[neg,nz,zero]) or member(compxb,'[pos,pz,zero]) then
			0
		elseif __b = 'inf then
		(
			(1 + signum(__x - __a) - unit_spike(__x - __a))/2
		)
		elseif __a = 'minf then
		(
			(1 - signum(__x - __b) - unit_spike(__x - __b))/2
		)
		else
			(simpfuncall('between, __x, __a, __b, first(__option)))
	elseif first(__option) = 'rclosed then
		if (compxa:sign(__x - __a)) = 'pos and member(compxb:sign(__x - __b),'[neg,nz,zero]) then
			1
		elseif member(compxa, '[neg,nz,zero]) or compxb = 'pos then
			0
		elseif __b = 'inf then
		(
			(1 + signum(__x - __a) - unit_spike(__x-__a))/2
		)
		elseif __a = 'minf then
		(
			(1 - signum(__x - __b) + unit_spike(__x - __b))/2
		)
		else
			(simpfuncall('between, __x, __a, __b, first(__option)))
	elseif first(__option) = 'lclosed then
		if member(compxa:sign(__x - __a), '[pos,pz,zero]) and (compxb:sign(__x - __b)) = 'neg then
			1
		elseif compxa='neg or member(compxb, '[pos,pz,zero]) then
			0
		elseif __b = 'inf then
		(
			(1 + signum(__x - __a) + unit_spike(__x - __a))/2
		)
		elseif __a = 'minf then
		(
			(1 - signum(__x - __b) - unit_spike(__x - __b))/2
		)
		else
			(simpfuncall('between, __x, __a, __b, first(__option)))
	else
		(simpfuncall('between, __x, __a, __b, first(__option)))
)$

simplifying(iif, simpiif)$
simplifying(unit_spike, simpunitspike)$
simplifying(unit_pulse, simpunitpulse)$
simplifying(between, simpbetween)$
simplifying(pwdelta, simppwdelta)$
simplifying(diff_pwdelta, simpdiffpwdelta)$

linearfunc(__e, __x):=
block
(
	[inflag : true, ratprint:false, _a, _b, _retval],
	errcatch
	(
		_b:subst(0, __x, __e),
		_a:subst(1, __x, __e) - _b,
		if is(equal(radcan(_a*__x + _b-__e),0))=true then
			_retval : [_a, _b]
		else
			_retval : false
	),
	if %% = false or emptyp(%%) then
		false
	else
		_retval
)$

quadraticfunc(__e, __x):=
block
(
	[inflag : true, algebraic:true, ratprint:false, _p, _q, _r, _retval],
	errcatch
	(
		_p:subst(0, __x, __e),
		_q:subst(1, __x, __e),
		_r:subst(2, __x, __e),
		if is(equal(radcan((_r-2*_q+_p)/2*__x^2 + (4*_q-_r-3*_p)/2*__x + _p - __e),0))=true then
			_retval : radcan([(_r-2*_q+_p)/2, (4*_q-_r-3*_p)/2, _p])
		else
		   _retval : false
	),
	if %% = false or emptyp(%%) then
		false
	else
		_retval
)$

list_remove(__l,__n,__m):=
block
(
	[partswitch:true, _retval : "error occurred"],
	if listp(__l) then
	(
		errcatch
		(
			if __n > length(__l) then
				__retval : __l
			elseif length(__l) > __n+__m-2 and __m >= 0 then
			(
				if __n > 0 then
				(
					_retval : append(rest(__l,__n-1-length(__l)),rest(__l,__n+__m-1))
				)
			)
			elseif __m >= 0 then
			(
				if __n > 0 then
				(
					_retval : rest(__l,__n-1-length(__l))
				)
			)
		),
		if %% = [] then
			_retval : ["Something went wrong"]
	)
	else
		_retval : ["L is not a list"],
	_retval
)$


factor_signum:true;

/* pwint() only requires that the function be piecewise continuous except at a countable number of points.
 * pwint() does not compute principle values.
 */

pwint(__e, __x, [__v]):=
block
(
	[inflag : true, _f : __e, _retval, _funcs, _scnmp : false, _p, _args, _l, _funcs, _go : false],
	if mapatom(__x) then
	(
		if length(__v) = 2 then
			apply('pwdefint, flatten([_f, __x, __v]))
		elseif length(__v) = 0 then
		(

			if not freeof('signum, gatherargs(unitstep2signum(_f), 'signum)) then
			(
				_f : pwsimp(_f, __x),
				_f : zerospikes(_f),
				if freeof('signum, gatherargs(unitstep2signum(_f), 'signum)) then
				(
					_go : true
				)
			)
			else
			(
				_go : true
			),
			_args : gatherargs(_f, "^"),
			for _p in _args do (
				if is(equal(second(_p), 1/2)) = true then
					_scnmp : true
			),
			if _scnmp = true then _f : scanmap('factor, _f),
			_f : maxmin2iif(_f),
			_f : if2sum(_f),
			_f : iif2sum(_f),
			_f : abs2signum(_f),
			_f : hstep2signum(_f),
			_f : charfun2sum(_f),
			_f : simpunitstep(_f, __x),
			_f : unitstep2signum(_f),
			if member(factor_signum, ['true,'all]) then _f : simpsignum(_f),
			if factor_signum = 'all then _f : simpsignumargs(_f,__x),
			_f : zerospikes(_f),
			_f : expandwrt(_f, 'signum, 'pwdelta, 'diff_pwdelta, 'between, 'unit_pulse),
			if _go then
			(
				_retval : _pwint(_f, __x),
				if not freeof('integrate, _retval) then
					funmake(nounify('integrate), [__e, __x])
				else
					_retval : zeroconstterms(_retval, __x)
			)
			else
			(
				funmake(nounify('integrate), [__e, __x])
			)
		)
		else
		(
			error("Wrong number of arguments to pwint.")
		)
	)
	else
		error("Variable of integration must be a symbol.")
)$

_pwint(__e, __x):=
block
(
	[inflag : true, _sum : 0, _op, _q],
	_op : if not mapatom(__e) then op(__e) else false,
	if _op = "+" then
	(
		for _q in __e do (
			_sum : _sum + intterm(_q, __x)
		),
		_sum
	)
	elseif _op = "*" then
	(
		intterm(__e, __x)
	)
	elseif _op = "^" then
	(
		if safe_op(expandwrt(__e, 'signum, 'pwdelta, 'diff_pwdelta, 'between, 'unit_pulse)) # "^" then
			pwint((expandwrt(__e, 'signum, 'pwdelta, 'diff_pwdelta, 'between, 'unit_pulse)), __x)
		else
			intterm(__e, __x)
	)
	elseif _op = 'signum then
	(
		intterm(__e, __x)
	)
	elseif _op = 'unit_pulse then
	(
		intterm(__e, __x)
	)
	elseif _op = 'between then
	(
		intbetween(__e, __x)
	)
	elseif _op = 'pwdelta then
	(
		intpwdelta(__e, __x)
	)
	elseif _op = 'sum then
	(
		intterm(__e,__x)
	)
	elseif not freeof('sum, __e) and not emptyp(gatherargs(__e, 'sum))
		and length(gatherargs(__e, 'sum)) = 1  then
	(
		intterm(__e,__x)
	)
	elseif _op = 'diff_pwdelta then
	(
		intdiffpwdelta(__e, __x)
	)
	elseif freeof('signum, 'pwdelta, 'diff_pwdelta, 'between, 'unit_pulse, __e) then
		integrate(__e,__x)
	else
		funmake(nounify('integrate), [__e, __x])
)$

__pwint(__e, __x):=
block
(
	[inflag : true],
	__e : zerospikes(__e),
	__e : expandwrt(__e, 'signum, 'pwdelta, 'diff_pwdelta, 'between, 'unit_pulse),
	if freeof('signum, gatherargs(__e, 'signum)) then
		_pwint(__e, __x)
	else
		funmake(nounify('integrate), [__e, __x])
)$

___pwint(__e, __x):=
block
(
	[inflag : true],
	__e : abs2signum(__e),
	__e : zerospikes(__e),
	__e : expandwrt(__e, 'signum, 'pwdelta, 'diff_pwdelta, 'between, 'unit_pulse),
	if freeof('signum, gatherargs(__e, 'signum)) then
		_pwint(__e, __x)
	else
		funmake(nounify('integrate), [__e, __x])
)$

intterm(__e, __x):=
block
(
	[inflag : true, _t, _sum : 0, _tmpt, _tmp1, _tmp2, _t1, _t2, _errflag : false, _op],
	_op : if not mapatom(__e) then op(__e) else false,
	if _op = "*" then
	(
		_t1 : xreduce("*", sublist(args(__e), lambda([_z], freeof(__x, _z)))),
		_t2 : xreduce("*", sublist(args(__e), lambda([_z], not freeof(__x, _z))))
	)
	elseif _op = "^" then
	(
		_t1 : _t2 : 1,
		if freeof(__x, __e) then
			_t1 : __e
		else
			_t2 : __e
	)
	elseif _op = "+" then
	(
		print("Not a term")
	)
	elseif not freeof('sum, __e) then
	(
		if safe_op(__e) = "*" then
		(
			_t1 : xreduce("*", sublist(args(__e), lambda([_z], freeof(__x, _z)))),
			_t2 : xreduce("*", sublist(args(__e), lambda([_z], not freeof(__x, _z))))
		)
		elseif not freeof('sum, __e) and not emptyp(gatherargs(__e, 'sum))
			and length(gatherargs(__e, 'sum)) = 1 then
		(
			_t1 : 1,
			_t2 : __e
		)
		else
			print("Not a Term")
	)
	else
	(
		_t1 : _t2 : 1,
		if freeof(__x, __e) then
			_t1 : __e
		else
			_t2 : __e
	),
	if not freeof('sum, _t2) then
	(
		_tmp1 : gatherargs(_t2, 'sum),
		if length(_tmp1) = 1 then
		(
			_tmp1 : first(_tmp1),
			_tmp2 : _t2/apply('sum, [first(_tmp1), second(_tmp1), third(_tmp1), fourth(_tmp1)]),
			_tmpt : pwint(first(_tmp1)*_tmp2,__x),
			_t1*apply('sum, [_tmpt, second(_tmp1), third(_tmp1), fourth(_tmp1)])
		)
		else
		(
			funmake(nounify('integrate), [__e, __x])
		)
	)
	elseif not freeof('pwdelta, _t2) then
		_t1*intpwdelta(_t2, __x)
	elseif not freeof('diff_pwdelta, _t2) then
		_t1*intdiffpwdelta(_t2, __x)
	elseif not freeof('between, _t2) then
	(
		_t : betweencheck(_t2),
		if _t = false then
			funmake(nounify('integrate), [__e, __x])
		else
			_t1*intbetween(_t, __x)
	)
	elseif not freeof('unit_pulse, _t2) then
	(
		_t : pulsecheck(_t2),
		if _t = false then
			funmake(nounify('integrate), [__e, __x])
		else
			_t1*intpulse(_t, __x)
	)
	elseif not freeof('signum, _t2) then
	(
		_t : signumcheck(_t2),
		if _t = false then
		(
			_tmpt : (pwsimp(_t2, __x, 'signum)),
			_t : signumcheck(_tmpt),
			if _t = 'false then
				integrate(_t1 * _tmpt, __x)
			else
			(
				if freeof('signum, _t) then
					_t1 * integrate(_t, __x)
				else
				(
					if safe_op(_t) = 'signum or safe_op(_t) = "*" or safe_op(_t) = "^" then
					(
						_t1 * intsignum(_t, __x)
					)
					else
						_t1 * pwint(_t, __x)
				)
			)
		)
		else
		(
			if freeof('signum, _t) then
				_t1*integrate(_t, __x)
			else
			(
				if safe_op(_t) = 'signum or safe_op(_t) = "*" or safe_op(_t) = "^" then
				(
					(_t1 * intsignum(_t, __x))
				)
				else
					_t1 * pwint(_t, __x)
			)
		)
	)
	elseif freeof('signum, 'pwdelta, 'diff_pwdelta, 'between, 'unit_pulse, __e) then
	(
		integrate(__e,__x)
	)
	else
		funmake(nounify('integrate), [__e, __x])
)$



intpwdelta(__term, __x):=
block
(
	[inflag:true, _a, _s, _ans, _lans, _z],
	_a : block([_s], _s:gatherargs(__term, 'pwdelta), _s),
	_a : flatten(_a),
	_s : pwdelta(first(_a)),
	if listp((_lans : linearfunc(first(_a), __x))) and listp((linearfunc(__term, _s))) then
	(
		_ans : -second(_lans)/first(_lans),
		_z : at(1/first(_lans)*__term/_s, [__x = _ans]) * (signum(first(_a)) + 1) / 2
	)
	else
		funmake(nounify('integrate), [__term, __x])
)$

intdiffpwdelta(__term, __x):=
block
(
	[inflag:true, _a, _s, _ans, _lans, _z],
	_a : block([_s], _s:gatherargs(__term, 'diff_pwdelta), _s),
	_s : diff_pwdelta(first(first(_a)), second(first(_a))),
	if listp((_lans : linearfunc(second(first(_a)), __x))) and nonnegintegerp(first(first(_a))) and listp((linearfunc(__term, _s))) then
	(
		_ans : -second(_lans)/first(_lans),
		_z: at(diff(((-1)^(first(first(_a)))*__term/_s),__x, first(first(_a))), [__x = _ans])/first(_lans)^(first(first(_a))+1) *
			(signum(second(first(_a))) + 1) / 2
	)
	else
		funmake(nounify('integrate), [__term, __x])
)$

intsignum(__term, __x):=
block
(
	[inflag : true, _a, _z, __retval : 0, _s, _t, _ans, _tmp],
	_a : flatten(listify(setify(gatherargs(__term, 'signum)))),
	_s : signum(first(_a)),
	if listp(_ans : linearfunc((first(_a)), __x)) then
	(
		_ans : -second(_ans)/first(_ans),
		if not mapatom(__term) and op(__term) = 'signum then
		(
			_z : subst(1, _s, __term),
			_z : integrate(_z, __x),
			_t : errcatch(_z : (_z - at(_z,[__x=_ans])) * _s),
			if _t = [] then
				_z : _z * _s
		)
		else
		(
			_z : subst(1, _s, __term),
			if freeof('abs, 'signum, 'between, _z) then
			(
				_z : integrate(_z,__x)
			)
			else
			(
				_z : (pwint(_z, __x))
			),
			_t : errcatch(_z : (_z - at(_z,[__x=_ans])) * _s),
			if _t = [] then
				_z : _z * _s
		),
		_z
	)
	elseif freeof('abs, 'signum, 'pwdelta, 'diff_pwdelta, 'between, __term) then
		integrate(__term, __x)
	else
		funmake(nounify('integrate), [__term, __x])
)$

intpulse(__term, __x):=
block
(
	[inflag : true, _x, _a, _b, _o, _z, _t, _ans_a, _ans_b, _op, _l],
	between2unitpulse(intbetween(zerospikes(unitpulse2between(__term)),__x))
)$

intbetween(__term, __x):=
block
(
	[inflag : true, _x, _a, _b, _o, _z, _t, _ans_a, _ans_b, _op, _l],
	_op : if not mapatom(__term) then op(__term) else false,
 	_l : listify(setify(gatherargs(__term, 'between))),
	_l : first(_l),	_x : first(_l),	_a : second(_l), _b : third(_l),
	_t : between(_x, _a, _b, fourth(_l)),
	if length(_l) > 0 and listp(_ans_a : linearfunc(_x - _a, __x))	and listp(_ans_b : linearfunc(_x - _b, __x)) then
	(
		_ans_a : -second(_ans_a)/first(_ans_a),
		_ans_b : -second(_ans_b)/first(_ans_b),
		if _op = 'between then
		(
			_z : __term/_t,
			_z : integrate(_z, __x),
			_z : _z * between(_x,_a,_b) +
				(signum(_b-_a) + 1)/2 * (at(_z, [__x=_ans_b]) * (1 + signum(_x - _b))/2	+ at(_z, [__x=_ans_a]) * (1 - signum(_x - _a))/2)
		)
		elseif member(_t, args(__term)) then
		(
			_z : __term/_t,
			if freeof('abs, 'signum, 'between, _z) then
			(
				_z : integrate(_z, __x)
			)
			else
			(
				_z : pwint(_z, __x)
			),
			_z : _z * between(_x,_a,_b) +
				(signum(_b-_a) + 1)/2 * (at(_z, [__x = _ans_b]) * (1 + signum(_x - _b))/2 + at(_z, [__x = _ans_a]) * (1 - signum(_x - _a))/2)
		)
		else
			_z : funmake(nounify('integrate), [__term, __x]),
		_z
	)
	elseif freeof('abs, 'signum, 'pwdelta, 'diff_pwdelta, 'between, __term) then
		integrate(__term, __x)
	else
		funmake(nounify('integrate), [__term, __x])
)$

linearize(__e):=
block
(
	[inflag:true, _f : __e],
	if not mapatom(__e) and op(__e) = "+" then
		xreduce("+", map(lambda([_z], linearize(_z)), inpargs(__e)))
	elseif not mapatom(__e) and op(__e) = "*" then
		betweencheck(signumcheck(__e))
	elseif not mapatom(__e) and op(__e) = "^" then
		betweencheck(signumcheck(__e))
	else
		_f,
	if not freeof('false, %%) then
		_f
	else
		%%
)$


linearizesignum(__e):=
block
(
	[inflag:true, __retval, _lk, _sum : 0, _a, _p],
	__e : expandwrt(__e, 'signum),
	if not mapatom(__e) and op(__e) = "+" then
	(
		for _p in __e do (
			_sum : _sum + linearizesignum(_p)
		),
		_sum
	)
	else
	(
		__retval : __e,
		_a : gatherargs(__e, "^"),
		map(lambda([_lk],
			if not mapatom(first(_lk)) and op(first(_lk)) = 'signum and oddp(second(_lk)) then
				__retval : subst(first(_lk), first(_lk)^second(_lk), __retval)
			elseif not mapatom(first(_lk)) and op(first(_lk)) = 'signum and evenp(second(_lk)) then
				__retval : subst(1-unit_spike(first(args(first(_lk)))), first(_lk)^second(_lk), __retval)
			),
			_a
		),
		__retval
	)
);

signumcheck(__e):=
block
(
	[inflag:true,__retval, _lk, _e, _q, _args, _dv, _go : true, _sum : 0, _a],
	if not mapatom(__e) and op(__e) = "+" then
	(
		for _p in __e do (
			_sum : _sum + signumcheck(_p)
		),
		if freeof('false, _sum) then
			_sum
		else
			'false
	)
	else
	(
		__retval : __e,
		_a : gatherargs(__e, "^"),
		map(lambda([_lk],
			if not mapatom(first(_lk)) and op(first(_lk)) = 'signum and oddp(second(_lk)) then
				__retval : subst(first(_lk), first(_lk)^second(_lk), __retval)
			elseif not mapatom(first(_lk)) and op(first(_lk)) = 'signum and evenp(second(_lk)) then
				__retval : subst(1-unit_spike(_lk), first(_lk)^second(_lk), __retval)
			),
			_a
		),
		_args : block([_s], _s:gatherargs(__retval, 'signum), flatten(listify(setify(_s)))),
		if not freeofall(['signum, 'unit_spike], _args) then
		(
			__retval : false,
			_args : []  /* I don't handle this kind of expression so return false */
		),
		while length(_args) > 0 do
		(
			_q:first(_args),
			_args:rest(_args),
			_dv : signum(_q),
			if not listp(linearfunc(__retval, _dv)) then
			(
				__retval : false,	 /* I don't handle this kind of expression so return false */
				_args : []
			)
		),
		__retval
	)
)$

betweencheck(__e):=
block
(
	[inflag:true,__retval, _lk, _e, _q, _args, _dv, _go : true, _sum : 0, _a],
	if not mapatom(__e) and op(__e) = "+" then
	(
		for _p in __e do (
			_sum : _sum + betweencheck(_p)
		),
		if freeof('false, _sum) then
			_sum
		else
			'false
	)
	else
	(
		__retval : __e,
		_a : gatherargs(__e, "^"),
		map(lambda([_lk],
			if not mapatom(first(_lk)) and op(first(_lk)) = 'between and second(_lk) > 0 then
				__retval : subst(first(_lk), first(_lk)^second(_lk), __retval)
			),
			_a
		),
		_args : block([_s], _s:gatherargs(__retval, 'between), listify(setify(_s))),
		while length(_args) > 0 do
		(
			_q:first(_args),
			_args:rest(_args),
			_dv : apply('between, _q),
			if not listp(linearfunc(__retval, _dv)) then
			(
				__retval : false,	 /* I don't handle this kind of expression so return false */
				_args : []
			)
		),
		__retval
	)
)$

pulsecheck(__e):=
block
(
	[inflag:true,__retval, _lk, _e, _q, _args, _dv, _go : true, _sum : 0, _a],
	if not mapatom(__e) and op(__e) = "+" then
	(
		for _p in __e do (
			_sum : _sum + pulsecheck(_p)
		),
		if freeof('false, _sum) then
			_sum
		else
			'false
	)
	else
	(
		__retval : __e,
		_a : gatherargs(__e, "^"),
		map(lambda([_lk],
			if not mapatom(first(_lk)) and op(first(_lk)) = 'unit_pulse and second(_lk)>0 then
				__retval : subst(first(_lk), first(_lk)^second(_lk), __retval)
			),
			_a
		),
		_args : block([_s], _s:gatherargs(__retval, 'unit_pulse), flatten(listify(setify(_s)))),
		if not freeof('signum, 'unit_spike, _args) then
		(
			__retval : false,
			_args : []  /* I don't handle this kind of expression so return false */
		),
		while length(_args) > 0 do
		(
			_q:first(_args),
			_args:rest(_args),
			_dv : unit_pulse(_q),
			if not listp(linearfunc(__retval, _dv)) then
			(
				__retval : false,	 /* I don't handle this kind of expression so return false */
				_args : []
			)
		),
		__retval
	)
)$

/* pwsimp() has respect for boundary conditions now, but it does not handle the pwdelta() functions */
pwsimp(__e, __x, [__options]) :=
block(
	[partswitch : true, use_between : false, inflag:true, opsubst : true, __retval : 0, _q, _t1, _t2, _sort_failed : false],
	_f : between2iif(__e),
	_f : maxmin2iif(_f),
	errcatch(_f : ifthen2if(_f)),
	_f : if2sum(_f),
	_f : iif2sum(_f),
	_f : charfun2sum(_f),
	__options : flatten(__options),
	_f : simppwdeltas(_f),
	if not freeof('unit_step, _f) then _f : scanmap(lambda([_z], simpunitstep(_z, __x)), _f),
	_f : convertall2signum(_f),
	_f : unitspike2signum(_f),
	if global = false then _f : expandwrt(_f, 'signum, 'unit_spike),
	if global = true then
		__retval : _nsignumsimp(_f, __x, __options)
	else
	(
		if not mapatom(_f) and op(_f) = "+" then
		(
			for _q in args(_f) do (
				__retval : __retval + _nsignumsimp(_q, __x, __options)
			)
		)
		elseif not mapatom(_f) and op(_f) = "*" then
		(
			_t1 : xreduce("*", sublist(args(_f), lambda([z], freeof(__x, z)))),
			_t2 : xreduce("*", sublist(args(_f), lambda([z], not freeof(__x, z)))),
			__retval : _t1 * _nsignumsimp(_t2, __x, __options)
		)
		else
		(
			__retval : _nsignumsimp(_f, __x, __options)
		)
	),
	if _sort_failed = true then
		__e
	else
		__retval
)$

_nsignumsimp(__e, __x, __options):=
block(
	[inflag:true, use_between : false, _q, _ans : [], _args, _args2:[], _yesno, _lkq, _p1, _p2, _t, _tmp, _tmplist, _nothing, _done, _l, _anst, _f : __e],
	__options : flatten(__options),
	_args:block([_s], _s:gatherargs(_f, 'signum), _s : listify(setify(_s)), _s),
	if not emptyp(_args) then
	(
		_t : [['minf, __x-'minf]],
		while not emptyp(_args) do (
			_q : first(_args),
			_args : rest(_args),
			_lkq : quadraticfunc(first(_q),__x),
			if listp(_lkq) then
			(
				if is(equal(first(_lkq),0))=true and is(equal(second(_lkq),0))=true then
				(
					/* no roots */ 6+9
				)
				elseif is(equal(first(_lkq),0))=true then
				(
					_temp : cons(-third(_lkq)/second(_lkq), _q),
					_t : cons(_temp, _t)
				)
				elseif is(equal(second(_lkq)^2-4*first(_lkq)*third(_lkq),0))=true then
				(
					_temp : cons(-second(_lkq)/(2*first(_lkq)), _q),
					_t : cons(_temp, _t)
				)
				elseif is(second(_lkq)^2-4*first(_lkq)*third(_lkq) > 0)=true then
				(
					_temp : cons((-second(_lkq)+sqrt(second(_lkq)^2-4*first(_lkq)*third(_lkq)))/(2*first(_lkq)), _q),
					_t : cons(_temp, _t),
					_temp : cons((-second(_lkq)-sqrt(second(_lkq)^2-4*first(_lkq)*third(_lkq)))/(2*first(_lkq)), _q),
					_t : cons(_temp, _t)
				)
				elseif is(second(_lkq)^2-4*first(_lkq)*third(_lkq) < 0)=true then
				(
					/* don't do anything */ 6+9
				)
				else
				(
					/* don't do anything */ 6+9
				)
			)
			else
			(
				/* don't do anything */ 6+9
			)
		),
		_t : cons(['inf, __x-'inf], _t),
		if not emptyp(_t) and length(_t) > 1 then
		(
			_t:sort(_t,lambda([c1,c2], block([t], t:errcatch(is(is(first(c1)<first(c2))='true)),if emptyp(t) then 'false else first(%%)))),
			_tmp:false,
			for i : 1 thru length(_t)-1 do if is(first(_t[i]) <= first(_t[i+1]))='unknown then (_tmp : true, _sort_failed : true),
			if equal(_tmp,true) then
			(
				_ans:__e
			)
			else
			(
				_tmplist:[],
				while not emptyp(_t) and length(_t) > 1 do (
					_tmp:_f,
					_p1:first(_t),
					_p2:second(_t),
					_t:rest(_t),
					_tmp : simp_given(_tmp, first(_p1) < __x, __x < first(_p2)),
					errcatch(_anst : if not member(first(_p1), ['minf,'inf]) then sublis([__x=first(_p1)], ratsimp(_f))),
					if %% = [] then _anst : 'und,
					_tmplist : flatten(cons([_tmp, first(_p1), _anst, first(_p1)],_tmplist))
				),
				_tmplist: flatten(cons([first(_p2), if not member(first(_p2), ['minf,'inf]) then
					sublis([__x=first(_p2)], ratsimp(_f)) else 'ind, first(_p2)],_tmplist)),
				if first(_tmplist) = 'inf and third(_tmplist) = 'inf then
					_tmplist : rest(_tmplist,2),
				_tmplist : reverse(_tmplist),
				if first(_tmplist) = 'minf and third(_tmplist) = 'minf then
					_tmplist : rest(_tmplist,2),
				for i : 1 thru length(__options) do
				(
					if member(__options[i], ['connect])=true then
					(
						__options : list_remove(__options, i, 1),
						i : length(__options)+100^100
					)
				),
				for i : 1 thru length(__options) do
				(
					if member(__options[i], ['closed, 'open, 'lclosed, 'rclosed, 'halfopen])=true then
					(
						__options : list_remove(__options, i, 1),
						i : i - 1
					)
				),
				if member('list,__options) then
					_ans: _tmplist
				elseif member('iif,__options) then
					_ans : simp_iif(pw(_tmplist, __x, cons('open, __options)))
				else
					_ans : simpspikes(pw(_tmplist, __x, cons('open, __options)))
			)
		)
		else
			_ans:__e
	)
	else
		_ans:__e,
	_ans
)$

pw(__L, __x, [__options]):=
block
(
    [__M, __k, _options:[], __rslt, __ans : [], _ls, _rs, _c],

	__options:flatten(__options),

	_ls:"<",
	_rs:">",
	if member('array, __options) then
		_options:cons('array, _options)
	elseif member('between, __options) then
	(
		_options:cons('between, _options)
	)
	elseif member('signum, __options) then
		_options:cons('signum, _options),
	if member('lclosed, __options) then
	(
		_ls: "[",
		_rs: ")",
		_options:cons('lclosed, _options)
	)
	elseif member('rclosed, __options) then
	(
		_ls: "(",
		_rs: "]",
		_options:cons('rclosed, _options)
	)
	elseif member('open, __options) then
	(
		_ls: "(",
		_rs: ")",
		_options:cons('open, _options)
	)
	elseif member('closed, __options) then
	(
		_ls: "[",
		_rs: "]",
		_options:cons('closed, _options)
	)
	elseif member('halfopen, __options) then
	(
		_options:cons('halfopen, _options)
	)
	else
		false,

	if listp(__L) and not emptyp(__L) then
	(
		if listp(__L[1]) and length(__L[1]) = 3 then
	    (
			if member('connect, __options) and __L[1][1] # 'minf then
				__L[1][3] : __L[1][3] - limit(__L[1][3], __x, __L[1][1], 'plus),
			__M:matrix(["If", __x, "in", %if(is(equal(__L[1][1], __L[1][2])), "[", %if(is(equal(__L[1][1],'minf)), "(", _ls)), __L[1][1], ",",
				__L[1][2], %if(is(equal(__L[1][1], __L[1][2])), "]", %if(is(equal(__L[1][2],'inf)), ")", _rs)), "then", __L[1][3]]),
			for __k: 2 thru length(__L) do (
				if listp(__L[__k]) then
					if length(__L[__k])=3 then
					(
						__L[__k+1][3] : __L[__k+1][3] +
						if member('connect, __options) then
							limit(__L[__k][3],__x,__L[__k][2],'minus) - limit(__L[__k+1][3],__x,__L[__k+1][1],'plus)
						else
							0,
						__M:addrow(__M,["If", __x, "in", %if(is(equal(__L[__k][1], __L[__k][2])), "[", _ls), __L[__k][1], ",", __L[__k][2],
							%if(is(equal(__L[__k][1], __L[__k][2])), "]", %if(is(equal(__L[__k][2],'inf)), ")", _rs)), "then", __L[__k][3]])
					)
			)
	    )
		elseif length(__L) > 2 and oddp(length(__L)) then
		(
			if member('connect, __options) and __L[1] # 'minf then
				__L[2] : __L[2] - limit(__L[2], __x, __L[1], 'plus),
			__M:matrix(["If", __x, "in", %if(is(equal(__L[1], __L[3])), "[", %if(is(equal(__L[1],'minf)), "(", _ls)), __L[1], ",", __L[3],
				%if(is(equal(__L[1], __L[3])), "]", _rs), "then", __L[2]]),
			for __k:2 thru length(__L)-3 step 2 do
			(
				if member('connect, __options) then
					__L[__k+2] : __L[__k+2] + limit(__L[__k], __x, __L[__k+1], 'minus) - limit(__L[__k+2], __x, __L[__k+1], 'plus),
				__M:addrow(__M,["If", __x, "in", %if(is(equal(__L[__k+1], __L[__k+3])), "[", _ls), __L[__k+1], ",", __L[__k+3],
					%if(is(equal(__L[__k+1], __L[__k+3])), "]", %if(is(equal(__L[__k+3],'inf)), ")", _rs)), "then", __L[__k+2]])
			)
		)
		elseif length(__L) = 1 then
			__M:matrix(["If", __x, "in", "(", minf, ",", inf, ")", "then", __L[1]])
		else
			error("Error : List has an even number of terms and it is the simple type."),
		if __ans # 'inconsistent then
		(
			if member('array, _options) then
			(
				disp(" "),
				disp(__M),
				__ans : "Done"
			)
			elseif member('ifthen, __options) then
				__ans : iif2ifthen(pwiifify(__M, _options))
			elseif member('iif, __options) then
				__ans : simp_iif(pwiifify(__M, _options))
			elseif member('%if, __options) then
				__ans : pw%ifify(__M, _options)
			elseif member('between, __options) then
				__ans : pwsumifybetween(__M, _options)
			elseif member('pulse, __options) then
				__ans : between2unitpulse(pwsumifybetween(__M, _options))
			elseif member('signum, __options) then
				__ans : pwsumifysignum(__M, _options)
			elseif member('unitstep, __options) then
				__ans : negunitstep(signum2unitstep(pwsumifysignum(__M, _options)),__x)
			elseif member('abs, __options) then
				__ans : signum2abs(pwsumifysignum(__M, _options))
			elseif use_between = 'true then
				__ans : pwsumifybetween(__M, _options)
			elseif use_between = 'false then
				__ans : pwsumifysignum(__M, _options)
			elseif use_between = 'pulse then
				__ans : between2unitpulse(pwsumifybetween(__M, _options))
			else
				__ans : pwsumifysignum(__M, _options)
		)
		else
			[__ans]
	)
	else
		error("First argument must be a list and not an empty list.")
)$

piecewise(__L,__x,[__options]):=pw(__L,__x,__options)$

pwsumifybetween(__M, [__options]):=block
(
    [__i,__retval:0, _option, _tmp:[]],

	__options:flatten(__options),

	if member('rclosed, __options) then
		(_option : 'rclosed)
	elseif member('open, __options) then
		(_option : 'open)
	elseif member('lclosed, __options) then
		(_option : 'lclosed)
	elseif member('closed, __options) then
		(_option : 'closed)
	elseif member('halfopen, __options) then
		(_option : 'halfopen)
	else
		(_option : 'none),

	/* inf * unit_spike(x-a) is NOT equivilent to pwdelta(x-a) */

    if matrixp(__M) and length(__M[1]) = 10 then
	(
		for __i: 1 thru length(__M) do (
			if is(equal(__M[__i,5],__M[__i,7]))=true and listp(__M[__i,10]) then
			(
				__retval : __retval +
				if length(__M[__i,10]) = 2 then
					second(__M[__i,10]) * diff_pwdelta(first(__M[__i,10]), __M[__i,2] - __M[__i,5])
				elseif length(__M[__i,10]) = 1 or (length(__M[__i,10]) = 2 and first(__M[__i,10]) = 0) then
					first(__M[__i,10]) * pwdelta(__M[__i,2] - __M[__i,5])
				else
				(
					_tmp : 'inconsistent
				)
			)
			elseif is(equal(__M[__i,5],__M[__i,7]))=true and not listp(__M[__i,10]) then
				__retval : __retval + __M[__i,10] * unit_spike(__M[__i,2] - __M[__i,5])
			elseif listp(__M[__i,10]) and is(equal(__M[__i,5],__M[__i,7]))=false then
				__retval : ['inconsistent]
			elseif not listp(__M[__i,10]) then
				if _option = 'none then
					__retval : __retval + __M[__i,10] * between(__M[__i,2], __M[__i,5], __M[__i,7])
				else
					__retval : __retval + __M[__i,10] * between(__M[__i,2], __M[__i,5], __M[__i,7], _option)

			else
			(
				__retval : ['inconsistent]
			)
		)
	)
	else
		error("This is not a user function."),
	if _tmp # 'inconsistent then
		__retval
	else
		[_tmp]
)$

pw%ifify(__M, [__options]):=block
(
    [__i,__retval:0, _option, _tmp:[]],

	__options:flatten(__options),

	if member('rclosed, __options) then
		(_option : 'rclosed)
	elseif member('open, __options) then
		(_option : 'open)
	elseif member('lclosed, __options) then
		(_option : 'lclosed)
	elseif member('closed, __options) then
		(_option : 'closed)
	elseif member('halfopen, __options) then
		(_option : 'halfopen)
	else
		(_option : 'none),

	/* inf * unit_spike(x-a) is NOT equivilent to pwdelta(x-a) */

    if matrixp(__M) and length(__M[1]) = 10 then
	(
		for __i: 1 thru length(__M) do (
			if is(equal(__M[__i,5],__M[__i,7]))=true and listp(__M[__i,10]) then
			(
				__retval : __retval +
				if length(__M[__i,10]) = 2 then
					second(__M[__i,10]) * diff_pwdelta(first(__M[__i,10]), __M[__i,2] - __M[__i,5])
				elseif length(__M[__i,10]) = 1 or (length(__M[__i,10]) = 2 and first(__M[__i,10]) = 0) then
					first(__M[__i,10]) * pwdelta(__M[__i,2] - __M[__i,5])
				else
				(
					_tmp : 'inconsistent
				)
			)
			elseif is(equal(__M[__i,5],__M[__i,7]))=true and not listp(__M[__i,10]) then
				__retval : __retval + __M[__i,10] * unit_spike(__M[__i,2] - __M[__i,5])
			elseif listp(__M[__i,10]) and is(equal(__M[__i,5],__M[__i,7]))=false then
				__retval : ['inconsistent]
			elseif not listp(__M[__i,10]) then
				if _option = 'closed then
					__retval : __retval + %if(__M[__i,2] >= __M[__i,5], %if(__M[__i,2] <= __M[__i,7], __M[__i,10], 0), 0)
				elseif _option = 'lclosed then
					__retval : __retval + %if(__M[__i,2] >= __M[__i,5], %if(__M[__i,2] < __M[__i,7], __M[__i,10], 0), 0)
				elseif _option = 'rclosed then
					__retval : __retval + %if(__M[__i,2] > __M[__i,5], %if(__M[__i,2] <= __M[__i,7], __M[__i,10], 0), 0)
				elseif _option = 'open then
					__retval : __retval + %if(__M[__i,2] > __M[__i,5], %if(__M[__i,2] < __M[__i,7], __M[__i,10], 0), 0)
				elseif _option = 'halfopen then
					__retval : __retval +
						%if(__M[__i,2] > __M[__i,5], %if(__M[__i,2] < __M[__i,7], __M[__i,10],
						%if(__M[__i,2] = __M[__i,7], __M[__i,10]/2, 0)), %if(__M[__i,2] = __M[__i,5], __M[__i,10]/2, 0))  /* this line tested okay 08-20-10 */
				else
					__retval : __retval +
						%if(__M[__i,2] > __M[__i,5], %if(__M[__i,2] < __M[__i,7], __M[__i,10],
						%if(__M[__i,2] = __M[__i,7], __M[__i,10]/2, 0)), %if(__M[__i,2] = __M[__i,5], __M[__i,10]/2, 0))  /* this line tested okay 08-20-10 */
			else
			(
				__retval : ['inconsistent]
			)
		)
	)
	else
		error("This is not a user function."),
	if _tmp # 'inconsistent then
		__retval
	else
		[_tmp]
)$

pwiifify(__M, [__options]):=block
(
    [__i,__retval:0, _option, _tmp:[]],

	__options:flatten(__options),

	if member('rclosed, __options) then
		(_option : 'rclosed)
	elseif member('open, __options) then
		(_option : 'open)
	elseif member('lclosed, __options) then
		(_option : 'lclosed)
	elseif member('closed, __options) then
		(_option : 'closed)
	elseif member('halfopen, __options) then
		(_option : 'halfopen)
	else
		(_option : 'none),

	/* inf * unit_spike(x-a) is NOT equivilent to pwdelta(x-a) */

    if matrixp(__M) and length(__M[1]) = 10 then
	(
		for __i: 1 thru length(__M) do (
			if is(equal(__M[__i,5],__M[__i,7]))=true and listp(__M[__i,10]) then
			(
				__retval : __retval +
				if length(__M[__i,10]) = 2 then
					second(__M[__i,10]) * diff_pwdelta(first(__M[__i,10]), __M[__i,2] - __M[__i,5])
				elseif length(__M[__i,10]) = 1 or (length(__M[__i,10]) = 2 and first(__M[__i,10]) = 0) then
					first(__M[__i,10]) * pwdelta(__M[__i,2] - __M[__i,5])
				else
				(
					_tmp : 'inconsistent
				)
			)
			elseif is(equal(__M[__i,5],__M[__i,7]))=true and not listp(__M[__i,10]) then
				__retval : __retval + __M[__i,10] * iif(equal(__M[__i,2],  __M[__i,5]), 1, 0)
			elseif listp(__M[__i,10]) and is(equal(__M[__i,5],__M[__i,7]))=false then
				__retval : ['inconsistent]
			elseif not listp(__M[__i,10]) then
				if _option = 'closed then
					__retval : __retval + iif(__M[__i,2] >= __M[__i,5], iif(__M[__i,2] <= __M[__i,7], __M[__i,10], 0), 0)
				elseif _option = 'lclosed then
					__retval : __retval + iif(__M[__i,2] >= __M[__i,5], iif(__M[__i,2] < __M[__i,7], __M[__i,10], 0), 0)
				elseif _option = 'rclosed then
					__retval : __retval + iif(__M[__i,2] > __M[__i,5], iif(__M[__i,2] <= __M[__i,7], __M[__i,10], 0), 0)
				elseif _option = 'open then
					__retval : __retval + iif(__M[__i,2] > __M[__i,5], iif(__M[__i,2] < __M[__i,7], __M[__i,10], 0), 0)
				elseif _option = 'halfopen then
					__retval : __retval +
						iif(__M[__i,2] > __M[__i,5], iif(__M[__i,2] < __M[__i,7], __M[__i,10],
						iif(equal(__M[__i,2], __M[__i,7]), __M[__i,10]/2, 0)), iif(equal(__M[__i,2], __M[__i,5]), __M[__i,10]/2, 0))
				else
					__retval : __retval +
						iif(__M[__i,2] > __M[__i,5], iif(__M[__i,2] < __M[__i,7], __M[__i,10],
						iif(equal(__M[__i,2], __M[__i,7]), __M[__i,10]/2, 0)), iif(equal(__M[__i,2], __M[__i,5]), __M[__i,10]/2, 0))
			else
			(
				__retval : ['inconsistent]
			)
		)
	)
	else
		error("This is not a user function."),
	if _tmp # 'inconsistent then
		__retval
	else
		[_tmp]
)$

pwsumifysignum(__M, [options]):=block
(
    [__i,__retval:0,__num1:0,__num2:0,_options, _tmp:[]],

	_options:flatten(options),

	if member('lclosed, _options) then
		(__num1:-1,__num2:-1)
	elseif member('open, _options) then
		(__num1:1,__num2:-1)
	elseif member('rclosed, _options) then
		(__num1:1,__num2:1)
	elseif member('closed, _options) then
		(__num1:-1,__num2:1)
	else
		(__num1:0,__num2:0),

	/* inf * unit_spike(x-a) is NOT equivilent to pwdelta(x-a) */

    if matrixp(__M) and length(__M[1]) = 10 then
	(
		for __i: 1 thru length(__M) do (
			if is(equal(__M[__i,5],__M[__i,7]))=true and listp(__M[__i,10]) then
				__retval : __retval +
			(
				if length(__M[__i,10]) = 2 then
					second(__M[__i,10]) * diff_pwdelta(first(__M[__i,10]), __M[__i,2] - __M[__i,5])
				elseif length(__M[__i,10]) = 1 or (length(__M[__i,10]) = 2 and first(__M[__i,10]) = 0) then
					first(__M[__i,10]) * pwdelta(__M[__i,2] - __M[__i,5])
				else
				(
					_tmp : 'inconsistent,
					0
				)
			)
			elseif is(equal(__M[__i,5],__M[__i,7]))=true and not listp(__M[__i,10]) then
				__retval : __retval + __M[__i,10] * unit_spike(__M[__i,2] - __M[__i,5])
			elseif listp(__M[__i,10]) and is(equal(__M[__i,5],__M[__i,7]))=false then
				__retval : ['inconsistent]
			elseif not listp(__M[__i,10]) then
				__retval : __retval + __M[__i,10] * (1+signum(__M[__i,7]-__M[__i,5]))/2*(signum(__M[__i,2] - __M[__i,5]) - signum(__M[__i,2] - __M[__i,7])
					+ signum(__M[__i,5] - __M[__i,7]) * __num1 * unit_spike(__M[__i,2] - __M[__i,5])
					- signum(__M[__i,5] - __M[__i,7]) * __num2 * unit_spike(__M[__i,2] - __M[__i,7]))/2
			else
			(
				__retval : ['inconsistent]
			)
		)
	)
	else
		error("This is not a user function."),
	if _tmp # 'inconsistent then
		__retval
	else
		[_tmp]
)$


pwifify(__M, [__options]):=block
(
    [__i, __retval, _L1:[], __num1:0, __num2:0, _options, _tmp:[]],
	_options : flatten(__options),
    if matrixp(__M) and length(__M[1]) = 10 then
	(
		for __i: 1 thru length(__M) do (
			if is(equal(__M[__i, 5],__M[__i, 7]))=true and listp(__M[__i, 10]) then
			(
				if length(__M[__i,10]) = 2 then
					_L1 : append(_L1, [equal(__M[__i,2],__M[__i,5])], [second(__M[__i,10]) * diff_pwdelta(first(__M[__i,10]), __M[__i,2] - __M[__i,5])])
				elseif length(__M[__i,10]) = 1 or (length(__M[__i,10]) = 2 and first(__M[__i,10]) = 0) then
					_L1 : append(_L1, [equal(__M[__i,2],__M[__i,5])], [first(__M[__i,10]) * pwdelta(__M[__i,2] - __M[__i,5])])
				else
				(
					_tmp : 'inconsistent
				)
			)
			elseif is(equal(__M[__i,5],__M[__i,7]))=true and not listp(__M[__i,10]) then
				_L1 : append(_L1,[equal(__M[__i,2],__M[__i,5])],[__M[__i,10] * unit_spike(__M[__i,2] - __M[__i,5])])
			elseif listp(__M[__i,10]) and is(equal(__M[__i,5],__M[__i,7]))=false then
				__retval : ['inconsistent]
			elseif not listp(__M[__i,10]) then
			(
				if member('lclosed, _options) then
				(
					_L1 : append(_L1, [between(__M[__i,2], __M[__i,5], __M[__i,7], 'lclosed)>0], [__M[__i,10]])
				)
				elseif member('open, _options) then
				(
					_L1 : append(_L1, [between(__M[__i,2], __M[__i,5], __M[__i,7], 'open)>0], [__M[__i,10]])
				)
				elseif member('rclosed, _options) then
				(
					_L1 : append(_L1, [between(__M[__i,2], __M[__i,5], __M[__i,7], 'rclosed)>0], [__M[__i,10]])
				)
				elseif member('closed, _options) then
				(
					_L1 : append(_L1, [between(__M[__i,2], __M[__i,5], __M[__i,7], 'closed)>0], [__M[__i,10]])
				)
				else
				(
					_L1 : append(_L1, [between(__M[__i,2], __M[__i,5], __M[__i,7])>0], [__M[__i,10]])
				)
			)
			else
			(
				__retval : ['inconsistent]
			)
		),
		_L1 : append(_L1, [true],[0]),
		__retval : apply("if", _L1)
	)
	else
		error("This is not a user function."),
	if _tmp # 'inconsistent then
		__retval
	else
		[_tmp]
)$

simpsignum(__e) :=
block(
	[opsubst : true, inflag : true],
	subst(
		'signum =
		lambda
		(
			[_z],
			block
			(
				[_s, _e, _t],
				_t : 1,
				_s : factor(_z),
				if not mapatom(_s) and op(_s) = "*" then
				(
					for _e in _s do
					(
						if not mapatom(_e) and op(_e) = "^" then
							_t : signum(first(first(gatherargs(_e,"^"))))^second(first(gatherargs(_e,"^")))*_t
						else
							_t : signum(_e)*_t
					)
				)
				elseif not mapatom(_s) and op(_s) = "^" then
					_t : signum(first(first(gatherargs(_s,"^"))))^second(first(gatherargs(_s,"^")))
				else
					_t : signum(_s),
			_t
			)
		),
		__e
	)
)$

simpsignumargs(__e, __x) :=
block
(
	[opsubst : true],
	subst('signum =
		lambda
		(
			[__s],
			block
			(
				[_t, _rv, _err, _start:1, _p],
				_t : linearfunc(__s, __x),
				if listp(_t) and is(first(_t) # 0) then
					_rv : signum(first(_t))*signum(__x+second(_t)/first(_t))
				else
				(
					_rv : signum(__s),
					_t : quadraticfunc(__s, __x),
					if listp(_t) and is(first(_t) # 0) and is(second(_t)^2-4*first(_t)*third(_t) > 0) then
					(
						_rv : signum(__x-(-second(_t)+sqrt(second(_t)^2-4*first(_t)*third(_t)))/(2*first(_t)))
							* signum(__x-(-second(_t)-sqrt(second(_t)^2-4*first(_t)*third(_t)))/(2*first(_t)))
					)
					elseif listp(_t) and is(first(_t) # 0) and is(equal(second(_t)^2-4*first(_t)*third(_t),0)) then
					(
						_rv : at(signum(__s),[__x=(-second(_t)+sqrt(second(_t)^2-4*first(_t)*third(_t)))/(2*first(_t))+1])-
							unit_spike(__x-(-second(_t))/(2*first(_t)))
					)
					elseif listp(_t) and is(first(_t) # 0) and is(second(_t)^2-4*first(_t)*third(_t)<0) = true then
					(
						_rv : at(signum(__s), __x=0)
					)
					else
					(
						_err : errcatch(_t : realroots(__s)),
						if _err # [] then
						(
							for _p in _t do (
								_start : signum(__x - rhs(_p)) * _start
							),
							_rv : _start
						)
						else
							_rv : signum(__s)
					)
				),
				_rv
			)
		),
		__e
	)
)$

negunitstep(__e, __x) :=
block
(
	[opsubst : true, inflag : true],
	subst('unit_step =
		lambda
		(
			[s],
			block
			(
				[_t],
				_t : linearfunc(s,__x),
				if listp(_t) and first(_t) < 0 then
				(
					1 - unit_step(-s) - unit_spike(-s)
				)
				else
					unit_step(s)
			)
		),
		__e
	)
)$

simpunitstep(__e, __x):=
block
(
	[inflag:true, _cnt : 0, _args:[], _ans, _roots1:[], _roots2:[], _maxval, _minval, __retval:0, _tmp],
	__e : expandwrt(__e, 'unit_step, 'signum, 'pwdelta, 'diff_pwdelta),
	if not mapatom(__e) and safe_op(__e) = "*" then
	(
		if length(gatherargs(__e, 'unit_step)) > 1 then
		(
			__retval : 1,
			for lk in __e do (
				if not mapatom(lk) and op(lk) = 'unit_step then
				(
					_args : first(gatherargs(lk, 'unit_step)),
					_ans : linearfunc(first(_args),__x),
					if listp(_ans) then
					(
						if first(_ans) > 0 then
						(
							_roots1 : cons(-second(_ans)/first(_ans), _roots1)
						)
						elseif first(_ans) < 0 then
						(
							_roots2 : cons(-second(_ans)/first(_ans), _roots2)
						)
						else
							__retval : __retval * lk
					)
					else
						__retval : __retval * lk
				)
				elseif not mapatom(lk) and op(lk)="^" then
				(
					_tmp : simpunitstep(lk,__x),
					if not mapatom(_tmp) and op(_tmp) = 'unit_step then
					(
						_args : first(gatherargs(_tmp, 'unit_step)),
						_ans : linearfunc(first(_args),__x),
						if listp(_ans) then
						(
							if first(_ans) > 0 then
							(
								_roots1 : cons(-second(_ans)/first(_ans), _roots1)
							)
							elseif first(_ans) < 0 then
							(
								_roots2 : cons(-second(_ans)/first(_ans), _roots2)
							)
							else
								__retval : __retval * lk
						)
						else
							__retval : __retval * lk
					)
					else
						__retval : __retval * lk
				)
				else
					__retval : __retval * lk
			),
			_maxval : apply('max, _roots1),
			_minval : apply('min, _roots2),
			__retval : __retval * unit_step(__x-_maxval) * unit_step(-__x+_minval)
		)
		elseif not mapatom(__e) and op(__e)="^" then
			__retval : simpunitstep(__e,__x)
		else
			__retval : __e
	)
	elseif not mapatom(__e) and safe_op(__e) = "+" then
	(
		for lk in __e do (
			__retval : simpunitstep(lk, __x) + __retval
		)
	)
	else
		__retval : __e,
	__retval
)$

periodic(__expr,__x,__a,__b):=
block
(
    if not is(__b>__a)=false then subst(__x - __a, __x, subst((__b - __a)*(__x/(__b - __a) - floor(__x/(__b - __a))) + __a, __x, __expr)) else error("a>b")
)$

intperiodic(__expr,__x,__a,__b):=
block
(
	if not is(__b>__a)=false then ((__x-__a)-(__b-__a )* ((__x-__a)/(__b-__a) - floor((__x-__a)/(__b-__a))))/(__b-__a) * pwint(__expr,__x,__a,__b)
		+ periodic(pwint(__expr,__x),__x,__a,__b) else error("a>b")
)$

pwdefint(__e, __x, __a, __b):=
block
(
	[_s, _ans, _c, _d, _p, _t, _done : false, algebraic : true],
	_s : pwint(__e, __x),
	if freeof(__x, __e) then
		_ans : __e * (__b - __a)
	elseif freeof('integrate, _s) and freeof('pwint, _s) then
	(
		_t : errcatch(_ans : at(_s,[__x=__b]) - at(_s,[__x=__a])),
		if _t = [] then
			_ans : funmake(nounify('integrate), [__e, __x, __a, __b])
		else
		(
			if freeof('und, 'inf, 'minf, _ans) then
			(
				_ans,
				_done : true
			),
			if not _done and not freeof('und, 'inf, 'minf, 'signum, 'between, _ans) then
			(
				_ans : ratsimp(_ans),
				if not freeof('und, 'inf, 'minf, _ans) then
				(
					_done : false
				)
				else
					_done : true
			),
			if not _done and (__b = 'inf or __a = 'minf) then
			(
				if __b = 'inf and __a = 'minf then
				(
					_ans : pwlimit(pwlimit(at(_s,[__x=_d]) - at(_s,[__x=_c]), _d, 'inf), _c, 'minf)
				)
				elseif __b = 'inf then
				(
					_ans : pwlimit(at(_s,[__x=_d]) - at(_s,[__x=__a]), _d, 'inf)
				)
				elseif __a = 'minf then
				(
					_ans : pwlimit(at(_s,[__x=__b]) - at(_s,[__x=_c]), _c, 'minf)
				)
				else
				(
					_ans
				)
			)
		)
	)
	else
		_ans : funmake(nounify('integrate), [__e, __x, __a, __b]),
	_ans
)$

/* This is the limit at inf or minf for any polynomial degree >= 0 argument to signum()
It is not needed most of the time and it also could be a lot better but alas I do not have time. */
pwlimit(__e, __x, __value):=
block
(
	[inflag : true, _tmp, __retval:0, _q, _args, _thelim],
	if is(__value = 'inf) = true or is(__value = 'minf) = true then
	(
		__retval : iif2sum(abs2iif(between2iif(__e))),
		_args : flatten(gatherargs(__retval, 'signum)),
		while length(_args) > 0 do
		(
			_q : first(_args),
			_args : rest(_args),
			if numberp(polydeg(_q,__x)) then
			(
				_thelim : polysignum(_q, __x, __value),
				__retval : subst(_thelim, signum(_q), __retval)
			)
		),
		__retval : facsum(__retval, __x),
		if safe_op(__retval) = "*" and not freeof('signum, __retval) then
		(
			_tmp : xreduce("*", sublist(args(__retval), lambda([_arg], freeof(__x, _arg)))),
			__retval : _tmp * limit(__retval / _tmp, __x, __value)
		)
		else
		(
			/* this does not always work, pwlimit in general needs more work but it is good enough for pw.mac as it is rarely called */
			__retval : apply('simp_given, cons(limit(__retval, __x, __value), map(lambda([_s], if __value = 'inf then __x > _s else __x < _s),
			listofvars(__retval))))
		)
	)
	else
		error("Third argument must be inf or minf."),
	__retval
)$

polysignum(__p, __x, __value):=
block
(
	[_p,_degree:polydeg(__p, __x),_sign_x],
	if (__value < 0) then _sign_x : -1 else _sign_x : 1,
	if numberp(_degree) and _degree # false then
	(
		if is(equal(_degree,0))=true then
			signum(__p)
		elseif is(oddp(_degree))=true then
		(
			_p:__p-ratcoeff(__p,__x,_degree)*__x^_degree,
			_sign_x * signum(ratcoeff(__p,__x,_degree)) + unit_spike(ratcoeff(__p,__x,_degree))*polysignum(_p,__x,__value)
		)
		elseif is(evenp(_degree))=true then
		(
			_p:__p-ratcoeff(__p,__x,_degree)*__x^_degree,
			signum(ratcoeff(__p,__x,_degree)) + unit_spike(ratcoeff(__p,__x,_degree))*polysignum(_p,__x,__value)
		)
		else
			signum(__p)
	)
	else
	(
		signum(__p)
	)
)$

/* get polynomial degree unless not a polynomial */
polydeg(__p,__x):=
block
(
	[_deg:-1,_okay,_ans],
	_okay : polynomialp(__p,[__x],
		lambda([_l], freeof(__x,_l)),
		lambda([_l], nonnegintegerp(_l))
	),
	if _okay then
		_deg : hipow(__p,__x)
	else
		false
)$

zerospikes(__e):=block([opsubst : true], subst('unit_spike = lambda([z], 0), __e))$
zeropwdeltas(__e):=block([opsubst : true], subst('pwdelta = lambda([z], 0), __e))$
zerodiffpwdeltas(__e):=block([opsubst : true], subst('diff_pwdelta = lambda([z1,z2], 0), __e))$
zeropwdeltas(__e):=zeropwdeltas(zerodiffpwdeltas(__e))$

convertall2signum(__ex) := block(
	__ex : maxmin2abs(__ex),
	__ex : abs2signum(__ex),
	__ex : unitpulse2signum(__ex),
	__ex : unitstep2signum(__ex),
	__ex : simpsignum(__ex)
)$


signum2abs(__e) :=
block(
	[__l],
	__e : simpsignum(__e),
	__l : flatten(gatherargs(__e,'signum)),
	for lk in __l do (
		__e : ratsubst(abs(lk)/lk, signum(lk),__e)),
  __e
)$

deltaint(__e, __x, [__v]):=
block(
	[inflag: true],
	if length(__v) = 0 then
		zerospikes(negunitstep(signum2unitstep(pwint(__e,__x)),__x))
	elseif length(__v) = 2 then
		apply('pwint, flatten([__e, __x, __v]))
	else
		print("wrong number of arguments to deltaint")
)$

zeroconstterms(__e, __x) :=
block
(
	[inflag : true],
	if (not mapatom(__e)) and (op(__e) = "+") then
		xreduce("+", sublist(args(__e), lambda([z], not freeof(__x, z))))
	else
		__e
)$

/* max2abs() and min2abs() are based on code in abs_integrate and are due to Barton Willis and are used with his permission */
max2abs(__e) := block([opsubst : true], subst('max = lambda([[__x]], xreduce(lambda([_%a,_%b], (_%a+_%b+abs(_%a-_%b))/2),__x)),__e))$
min2abs(__e) := block([opsubst : true], subst('min = lambda([[__x]], xreduce(lambda([_%a,_%b], (_%a+_%b-abs(_%a-_%b))/2),__x)),__e))$
maxmin2abs(__e):= max2abs(min2abs(__e))$

/* not exact */
between2signum(__e):= block([opsubst:true], subst('between = lambda([__x,__a,__b, __o], (1+signum(__b-__a))/2*(signum(__x-__a)-signum(__x-__b))/2), __e))$
between2unitstep(__e):= block([opsubst:true], subst('between = lambda([__x,__a,__b, __o], unit_step(__x-__a)-unit_step(__x-__b)), __e))$
charfun2signum(__e):= block([opsubst:true], subst('charfun2 = lambda([__x,__a,__b], (signum(__x-__a)-signum(__x-__b))/2), __e))$

/* exactly true transformations from here on.  Not to say everything before here is not exact. signum2abs() has a divide by zero problem, so there
is also signum2abs2() below which does not have this problem.  That advantage is gained at the price of not being able to express the answer in terms of the abs() function. 
zeroconstterms(), zerospikes(), signum2abs(), zeropwdeltas() and some diff()'s are not exact at singular points all the time which is by design */

between2unitpulse(__e):= block([opsubst : true],
	subst('between = lambda([__x,__a,__b, [__o]],
			block
			(
				[_n1, _n2],
				if member('lclosed, __o) then
					(_n1:1/2, _n2:-1/2)
				elseif member('open, __o) then
					(_n1:-1/2, _n2:-1/2)
				elseif member('rclosed, __o) then
					(_n1:-1/2, _n2:1/2)
				elseif member('closed, __o) then
					(_n1:1/2, _n2:1/2)
				else
					(_n1:0, _n2:0),
				(1+signum(__b-__a))/2 * (unit_pulse(__x/(__b-__a)+__a/(__a-__b)) + _n1 * unit_spike(__x - __a) + _n2 * unit_spike(__x - __b))
			)
		),
	__e
	)
)$


simppwdeltas(__e) :=
block(
	[inflag : true, _l, _m, _q, _p, lk, _sum : 0, _ans],
	__e : expandwrt(__e, 'pwdelta),
	if safe_op(__e) = "+" then
	(
		for _p in __e do (
			_sum : _sum + simppwdeltas(_p)
		),
		_sum
	)
	else
	(
		_m : gatherargs(__e, "^"),
		_l1 : flatten(gatherargs(__e, 'pwdelta)),
		for lk in _l1 do (
			__e : ratsubst(0, lk * pwdelta(lk),__e),
			__e : ratsubst(0, signum(lk) * pwdelta(lk),__e),
			__e : ratsubst(0, sin(lk) * pwdelta(lk),__e),
			__e : ratsubst(0, tan(lk) * pwdelta(lk),__e),
			while not emptyp(_m) do (
				_p : first(_m),
				_m : rest(_m),
				_q : second(_p),
				if _q > 0 then
				(
					__e : ratsubst(0, lk ^_q * pwdelta(lk),__e),
					__e : ratsubst(0, sin(lk) ^_q * pwdelta(lk),__e),
					__e : ratsubst(0, sin(lk ^_q) * pwdelta(lk),__e),
					__e : ratsubst(0, tan(lk) ^_q * pwdelta(lk),__e),
					__e : ratsubst(0, tan(lk ^_q) * pwdelta(lk),__e)
				)
			)
		),
		__e
	)
)$

simpspikes(__e) :=
block(
	[inflag : true, use_between:false, _l, _m, _q, _p, lk, _sum : 0, _done : false, _most1, _most2, _tmp, _tmp1, _stop, _ans],
	__e : expandwrt(__e, 'unit_spike),
	if safe_op(__e) = "+" then
	(
		for _p in __e do (
			_sum : _sum + simpspikes(_p)
		),
		_sum
	)
	else
	(
		_m : gatherargs(__e, "^"),
		for lk in _m do (
			if nonnegintegerp(second(lk)) and safe_op(first(lk)) = 'unit_spike then
				__e : ratsubst(first(lk), first(lk)^second(lk),__e)
		),
		_l : listify(setify(flatten(gatherargs(__e, 'unit_spike)))),
		for lk in _l do (
			__e : subst(0, lk * unit_spike(lk),__e),
			__e : subst(0, signum(lk) * unit_spike(lk),__e),
			__e : subst(0, sin(lk) * unit_spike(lk),__e),
			__e : subst(0, tan(lk) * unit_spike(lk),__e)
		),
		_l : listify(setify(flatten(gatherargs(__e, 'unit_spike)))),
		while not emptyp(_l) do (
			_p : first(_l),
			_l : rest(_l),
			_l2 : listify(setify(flatten(gatherargs(__e, 'unit_spike)))),
			while not emptyp(_l2) do (
				_q : first(_l2),
				_l2 : rest(_l2),
				if is(notequal(_p,_q))=true then
					__e : ratsubst(0, unit_spike(_p) * unit_spike(_q),__e)
			)
		),
		_l : flatten(gatherargs(__e, 'unit_spike)),
		if not emptyp(_l) then
		(
			_tmp : __e,
			while not emptyp(_l) do (
				_p : first(_l),
				_l : rest(_l),
				_tmp : subst(1, unit_spike(_p), _tmp)
			),
			_most1 : reverse(countvars(_tmp)),
			_l : flatten(gatherargs(__e, 'unit_spike)),
			_most2 : reverse(countvars(_l)),
			_stop : false,
			if not emptyp(_most1) and not emptyp(_most2) and is(equal(first(first(_most1)),first(first(_most2)))) then
			(
				_x : first(first(_most1))
			)
			elseif emptyp(_most1) and emptyp(_most2) then
			(
				_stop : true
			)
			else
			(
				_x : first(first(_most2))
			),
			if not _stop then
			(
				_expr : first(_l),
				if listp(_ans:linearfunc(_expr,_x)) then
				(
					if is(notequal(first(_ans),0))=true then
					(
						_ans : -second(_ans)/first(_ans),
						_tmp : subst(1, unit_spike(_expr), __e),
						_t : errcatch(_tmp1 : sublis([_x = _ans], ratsimp(_tmp))),
						if _t = [] then
						(
							__e
						)
						else
						(
						   __e : _tmp1  * unit_spike(_expr)
						)
					)
				)
			)
		),
		__e
	)
)$

/* needs more work */
if2sum(__e) := block(
	[opsubst : true, inflag:true],
	subst('%if =
		lambda(
			[__s,__a,__b],
			block
			(
				[_p:safe_op(__s),_q:safe_op(inpart(__s,1)),_r:safe_op(inpart(__s,2)),
				_a:rhs(inpart(__s,1)), _b:rhs(inpart(__s,2)), _x:lhs(inpart(__s,1)), _x1:lhs(inpart(__s,1)), _x2:lhs(inpart(__s,2))],
				if safe_op(__s) = ">" then
				(
					(signum(lhs(__s))+1-unit_spike(lhs(__s)))/2 * __a + (1-signum(lhs(__s))+unit_spike(lhs(__s)))/2 * __b
				)
				elseif safe_op(__s) = "=" then
				(
					unit_spike(lhs(__s)) * __a + (1-unit_spike(lhs(__s))) * __b
				)
				elseif safe_op(__s) = "#" then
				(
					unit_spike(lhs(__s)) * __b + (1-unit_spike(lhs(__s))) * __a
				)
				elseif apply(_p, [true,false,false]) = true
					and apply(_p, [false, false, false]) = false and _q = "=" and _r = ">" and _x1 = _x2 then
				(
					 __a * (signum(_x) + unit_spike(_x)+1)/2 + __b * (1-signum(_x)-unit_spike(_x))/2
				)
				elseif apply(_p, [true,false,false]) = true
					and apply(_p, [false, false, false]) = false and _q = ">" and _r = "=" and _x1 = _x2 then
				(
					(signum(_x) + unit_spike(_x)+1)/2 * __a + (1-signum(_x)-unit_spike(_x))/2 * __b
				)
				elseif apply(_p, [true,true,true]) = true and _x1 = _x2 and apply(_p, [false, true, true]) = false
					and _q = ">" and _r = "<" then
				(
					__a * between(_x,_a,_b,'open) + __b * (1-between(_x, _a, _b, 'open))
				)
				else
					%if(__s, __a, __b))),
		__e)
)$

pushoutiif(__e):=sum2iif(((iif2sum(__e))));

iif2sum(__e) := block(
	[opsubst : true, inflag:true,__s,__a,__b],
	subst('iif =
		lambda(
			[__s,__a,__b],
			block
			(
				[_op,_p,_q,_r,_a,_b,_x,_x1,_x2],
				_op:safe_op(__s),
				_p : lhs(__s),
				_q : rhs(__s),
				_p : _p - _q,
				if _op = ">" then
					__s : _p > 0
				elseif _op = "<" then
					__s : _p < 0
				elseif _op = ">=" then
					__s : _p >= 0
				elseif _op = "<=" then
					__s : _p <= 0,
				_p:safe_op(__s),
				_q:safe_op(inpart(__s,1)),
				_r:safe_op(inpart(__s,2)),
				_a:rhs(inpart(__s,1)),
				_b:rhs(inpart(__s,2)),
				_x:lhs(inpart(__s,1)),
				_x1:lhs(inpart(__s,1)),
				_x2:lhs(inpart(__s,2)),
				if safe_op(__s) = ">" then
					(signum(lhs(__s))+1-unit_spike(lhs(__s)))/2 * __a + (1-signum(lhs(__s))+unit_spike(lhs(__s)))/2 * __b
				elseif safe_op(__s) = "<" then
					(1-signum(lhs(__s))-unit_spike(lhs(__s)))/2 * __a + (signum(lhs(__s))+unit_spike(lhs(__s))+1)/2 * __b
				elseif safe_op(__s) = ">=" then
					(signum(lhs(__s))+1+unit_spike(lhs(__s)))/2 * __a + (1-signum(lhs(__s))-unit_spike(lhs(__s)))/2 * __b
				elseif safe_op(__s) = "<=" then
					(1-signum(lhs(__s))+unit_spike(lhs(__s)))/2 * __a + (signum(lhs(__s))-unit_spike(lhs(__s))+1)/2 * __b
				elseif safe_op(__s) = 'equal then
					unit_spike(first(args(__s))-second(args(__s))) * __a + (1-unit_spike(first(args(__s))-second(args(__s)))) * __b
				elseif safe_op(__s) = 'notequal then
					unit_spike(first(args(__s))-second(args(__s))) * __b + (1-unit_spike(first(args(__s))-second(args(__s)))) * __a
				else
					iif(__s, __a, __b))),
		__e)
)$

sum2iif(__e):= signum2iif((unitspike2iif(between2iif(unitpulse2between(__e)))));

/* pulliniif() (and especially pullinif()) are slow so I do not recommend their use for very complicated piecewise functions
	but they can be useful for extending integrate() for simple cases and simplification is other cases. */

pullinif(__e):=
block(
	[_done:false, _a, inflag:true],
	while not _done do (
	_a : _pull_in_if(__e),
	if _a = __e then
		_done : 'true,
		__e : _a
	),
	_a
)$

_pull_in_if(__e) :=
block(
	[opsubst : 'true, inflag : 'true, %a, %__b, _a, _b, _sum:0, _pod:1, _exp, _op, _args, _p],
	if mapatom(__e) then
		__e
	elseif safe_op(__e) = "+" and not freeof('%if, args(__e)) then
	(
		for _p in __e do (_sum : add_it_if(_sum, _pull_in_if(_p))),
		_sum
	)
	elseif safe_op(__e) = "*" and not freeof('%if, args(__e)) then
	(
		for _p in __e do (_pod : mult_it_if(_pod, _pull_in_if(_p))),
		_pod
	)
	elseif safe_op(__e) = "^" and not freeof('%if, args(__e)) then
	(
		_pull_in_if(exp_it_if(_pull_in_if(first(__e)),_pull_in_if(second(__e))))
	)
	elseif length(args(__e)) = 1 and not freeof('%if, args(__e)) then
	(
		_op : safe_op(__e),
		%a:_pull_in_if(first(args(__e))),
		if safe_op(%a) = '%if then
			apply('%if, [inpart(%a, 1), _pull_in_if(_op(_pull_in_if(inpart(%a,2)))), _pull_in_if(_op(_pull_in_if(inpart(%a,3))))])
		else
			_pull_in_if(_op(_pull_in_if(%a)))
	)
	elseif safe_op(__e) = '%if and not freeof('%if, args(__e)) then
	(
		%a:args(__e),
		apply('%if, [inpart(%a, 1), _pull_in_if(inpart(%a,2)), _pull_in_if(inpart(%a,3))])
	)
	elseif freeof('%if, args(__e)) then
		__e
	else
	(
		__e
	)
)$

add_it_if(%a,%b):=
block(
	[inflag:true],
	if not mapatom(%b) and op(%b) = '%if then
		%if(inpart(%b,1), _pull_in_if(%a+inpart(%b,2)), _pull_in_if(%a+inpart(%b,3)))
	else
	(
		if not mapatom(%a) and op(%a) = '%if then
			%if(inpart(%a,1), _pull_in_if(%b+inpart(%a,2)), _pull_in_if(%b+inpart(%a,3)))
		else
		(
			%a+%b
		)
	)
)$

mult_it_if(%a,%b):=
block(
	[inflag:true],
	if not mapatom(%b) and op(%b) = '%if then
		%if(inpart(%b,1), _pull_in_if(%a*inpart(%b,2)), _pull_in_if(%a*inpart(%b,3)))
	else
	(
		if not mapatom(%a) and op(%a) = '%if then
			%if(inpart(%a,1), _pull_in_if(%b*inpart(%a,2)), _pull_in_if(%b*inpart(%a,3)))
		else
		(
			%a*%b
		)
	)
)$

exp_it_if(%a,%b):=
block(
	[inflag:true],
	if not mapatom(%b) and op(%b) = '%if then
		%if(inpart(%b,1), _pull_in_if(%a^inpart(%b,2)), _pull_in_if(%a^inpart(%b,3)))
	else
	(
		if not mapatom(%a) and op(%a) = '%if then
			%if(inpart(%a,1), _pull_in_if(inpart(%a,2)^%b), _pull_in_if(inpart(%a,3)^%b))
		else
		(
			%a^%b
		)
	)
)$

pulliniif(__e):=
block(
	[_done:false, _a, inflag:true],
	while not _done do (
	_a : _pull_in_iif(__e),
	if _a = __e then
		_done : 'true,
		__e : _a
	),
	_a
)$

_pull_in_iif(__e) :=
block(
	[opsubst : 'true, inflag : 'true, %a, %b, _a, _b, _sum:0, _pod:1, _exp, _op, _args, _p],
	if mapatom(__e) then
		__e
	elseif safe_op(__e) = "+" and not freeof('iif, args(__e)) then
	(
		for _p in __e do (_sum : add_it_iif(_sum, _pull_in_iif(_p))),
		_sum
	)
	elseif safe_op(__e) = "*" and not freeof('iif, args(__e)) then
	(
		for _p in __e do (_pod : mult_it_iif(_pod, _pull_in_iif(_p))),
		_pod
	)
	elseif safe_op(__e) = "^" and not freeof('iif, args(__e)) then
	(
		_pull_in_iif(exp_it_iif(_pull_in_iif(first(__e)),_pull_in_iif(second(__e))))
	)
	elseif length(args(__e)) = 1 and not freeof('iif, args(__e)) then
	(
		_op : safe_op(__e),
		%a : _pull_in_iif(first(args(__e))),
		if safe_op(%a) = 'iif then
			apply('iif, [inpart(%a, 1), _pull_in_iif(_op(_pull_in_iif(inpart(%a,2)))), _pull_in_iif(_op(_pull_in_iif(inpart(%a,3))))])
		else
			_pull_in_iif(_op(_pull_in_iif(%a)))
	)
	elseif safe_op(__e) = 'iif and not freeof('iif, args(__e)) then
	(
		%a:args(__e),
		apply('iif, [inpart(%a, 1), _pull_in_iif(inpart(%a,2)), _pull_in_iif(inpart(%a,3))])
	)
	elseif freeof('iif, args(__e)) then
		__e
	else
	(
		__e
	)
)$

add_it_iif(%a,%b):=
block(
	[inflag:true],
	if not mapatom(%b) and op(%b) = 'iif then
		iif(inpart(%b,1), _pull_in_iif(%a+inpart(%b,2)), _pull_in_iif(%a+inpart(%b,3)))
	else
	(
		if not mapatom(%a) and op(%a) = 'iif then
			iif(inpart(%a,1), _pull_in_iif(%b+inpart(%a,2)), _pull_in_iif(%b+inpart(%a,3)))
		else
		(
			%a+%b
		)
	)
)$

mult_it_iif(%a,%b):=
block(
	[inflag:true],
	if not mapatom(%b) and op(%b) = 'iif then
		iif(inpart(%b,1), _pull_in_iif(%a*inpart(%b,2)), _pull_in_iif(%a*inpart(%b,3)))
	else
	(
		if not mapatom(%a) and op(%a) = 'iif then
			iif(inpart(%a,1), _pull_in_iif(%b*inpart(%a,2)), _pull_in_iif(%b*inpart(%a,3)))
		else
		(
			%a*%b
		)
	)
)$

exp_it_iif(%a,%b):=
block(
	[inflag:true],
	if not mapatom(%b) and op(%b) = 'iif then
		iif(inpart(%b,1), _pull_in_iif(%a^inpart(%b,2)), _pull_in_iif(%a^inpart(%b,3)))
	else
	(
		if not mapatom(%a) and op(%a) = 'iif then
			iif(inpart(%a,1), _pull_in_iif(inpart(%a,2)^%b), _pull_in_iif(inpart(%a,3)^%b))
		else
		(
			%a^%b
		)
	)
)$

if2iif(__e) := block(
	[opsubst : true, inflag:true],
	subst('%if =
		lambda(
			[__s,__a,__b],
			block
			(
				[_p:safe_op(__s),_q:safe_op(inpart(__s,1)),_r:safe_op(inpart(__s,2)),
				_a:rhs(inpart(__s,1)), _b:rhs(inpart(__s,2)), _x:lhs(inpart(__s,1)), _x1:lhs(inpart(__s,1)), _x2:lhs(inpart(__s,2))],
				if member(safe_op(__s), [">", ">=", "<", "<=", 'equal, 'notequal]) then
				(
					simp_iif(iif(__s, __a, __b))
				)
				elseif apply(_p, [true,false,false]) = true
					and apply(_p, [false, false, false]) = false and _q = "=" and _r = ">" and _x2 = _x1 and length(args(__s)) = 2 then
				(
						simp_iif(iif(_x >= _a, __a, __b))
				)
				elseif apply(_p, [true,false,false]) = true
					and apply(_p, [false, false, false]) = false and _q = ">" and _r = "=" and _x1 = _x2 and length(args(__s)) = 2 then
				(
					simp_iif(iif(_x >= _a, __a, __b))
				)
				elseif apply(_p, [true,true,true]) = true and apply(_p, [false, true, true]) = false
					and member(_q,   [">", ">=", "<", "<="]) and member(_r , [">", ">=", "<", "<="]) and _x1 = _x2 and length(args(__s)) = 2 then
				(
					simp_iif(iif(apply(_q, [_x, _a]), iif(apply(_r, [_x, _b]), __a, __b), __b))
				)
				else
					%if(__s, __a, __b)
			)
		),
	__e)
)$

signum2abs2(__e):=
block(
	[__l],
	__e : simpsignum(__e),
	__l : flatten(gatherargs(__e,'signum)),
	for lk in __l do (
		__e : ratsubst(abs(lk), lk*signum(lk),__e)),
  __e
)$

between2if(__e):=
block(
	[opsubst : true, inflag : true],
	subst('between =
		lambda([__x,__a,__b,__o],
		[_a],
		if __o = 'lclosed  then /* okay */
			%if(__x>=__a and __x<__b, 1, 0)
		elseif __o = 'closed then /* okay */
			%if(__x>=__a and __x<=__b, 1, 0)
		elseif __o = 'rclosed then /* okay */
			%if(__x>__a and __x<=__b, 1, 0)
		elseif __o = 'open then /* okay */
			%if(__x>__a and __x<__b, 1, 0)
		elseif __o = 'halfopen then /* okay */
			%if(__x>__a, %if(__x<__b, 1, %if(__x = __b, 1/2, 0)), %if(__x = __a, %if(__b>__a, 1/2, 0), 0))
		else /* okay */
			%if(__x>__a, %if(__x<__b, 1, %if(__x = __b, 1/2, 0)), %if(__x = __a, %if(__b>__a, 1/2, 0), 0))
		),
		__e)
)$

between2iif(__e):=
block(
	[opsubst : true, inflag : true],
	subst('between =
		lambda([__x,__a,__b,__o],
		[_a],
		if __o = 'lclosed  then /* okay */
			iif(__x>=__a, iif(__x<__b, 1, 0), 0)
		elseif __o = 'closed then /* okay */
			iif(__x>=__a, iif(__x<=__b, 1, 0), 0)
		elseif __o = 'rclosed then /* okay */
			iif(__x>__a, iif(__x<=__b, 1, 0), 0)
		elseif __o = 'open then /* okay */
			iif(__x>__a, iif(__x<__b, 1, 0), 0)
		elseif __o = 'halfopen then /* okay */
			iif(__x>__a, iif(__x<__b, 1, iif(equal(__x,__b), 1/2, 0)), iif(equal(__x,__a), iif(__b>__a, 1/2, 0), 0))
		else /* okay */
			iif(__x>__a, iif(__x<__b, 1, iif(equal(__x,__b), 1/2, 0)), iif(equal(__x,__a), iif(__b>__a, 1/2, 0), 0))
		),
		__e)
)$

unitpulse2between(__e):=block([opsubst : true], subst('unit_pulse = lambda([_s], between(_s,0,1)), __e))$
if2ifthen(__e) := block([opsubst : true], subst('%if = lambda([_cnd, __a, __b], if _cnd then __a else __b), __e))$
iif2ifthen(__e) := block([opsubst : true], subst('iif = lambda([_cnd, __a, __b], if _cnd then __a else __b), __e))$
unitspike2if(__e) := block([opsubst : true], subst('unit_spike = lambda([__s], %if(equal(__s,0), 1, 0)), __e))$
unitspike2iif(__e) := block([opsubst : true], subst('unit_spike = lambda([__s], iif(equal(__s,0), 1, 0)), __e))$
signum2if(__e) := block([opsubst : true, inflag : true], subst('signum = lambda([__s], %if(__s > 0, 1, %if(__s < 0, -1, 0))), __e))$
signum2iif(__e) := block([opsubst : true, inflag : true], subst('signum = lambda([__s], iif(__s > 0, 1, iif(__s < 0, -1, 0))), __e))$
abs2iif(__e) := block([opsubst : true, inflag : true], subst('abs = lambda([__s], %if(__s > 0, __s, -__s)), __e))$
abs2iif(__e) := block([opsubst : true, inflag : true], subst('abs = lambda([__s], iif(__s > 0, __s, -__s)), __e))$
unitstep2if(__e) := block([opsubst : true], subst('unit_step = lambda([__s], %if(__s > 0, 1, 0)), __e))$
unitstep2iif(__e) := block([opsubst : true], subst('unit_step = lambda([__s], iif(__s > 0, 1, 0)), __e))$
charfun2between(_e):= block([opsubst:true], subst('charfun2 = lambda([__x,__a,__b], between(__x,__a,__b, 'lclosed)), _e))$
charfun2sum(_e):= block([opsubst:true], subst('charfun2 = lambda([__x,__a,__b], iif2sum(between2iif(between(__x,__a,__b, 'lclosed)))), _e))$
unitspike2unitstep(__e) := block([opsubst : true], subst('unit_spike = lambda([__s], -unit_step (__s)+2*unit_step(-__s)*unit_step(__s)-unit_step(-__s)+1), __e))$
unitpulse2unitstep(__e) := block([opsubst : true], subst('unit_pulse = lambda([__s], (unit_step(__s)-unit_step(-__s)-unit_step(__s-1)+unit_step(1-__s))/2), __e))$
abs2unitstep(__e) := block([opsubst : true], subst('abs = lambda([__s], __s * unit_step(__s) - __s * unit_step(-__s)), __e))$
signum2unitstep(__e) := block([opsubst : true], subst('signum = lambda([__s], (unit_step(__s)-unit_step(-__s))), __e))$
unitpulse2signum(__e) := block([opsubst : true], subst('unit_pulse = lambda([__s], (signum(__s) - signum(__s - 1))/2), __e))$
abs2signum(__e) := block([opsubst : true], subst('abs = lambda([__s], __s * signum(__s)), __e));
unitspike2signum(__e) := block([opsubst : true], subst('unit_spike = lambda([__s], 1-signum(__s)^2), __e))$
unitstep2signum(__e) := block([opsubst : true], subst('unit_step = lambda([__s], (signum(__s)+1-unit_spike(__s))/2), __e))$
unitspike2signum(__e) := block([opsubst : true], subst('unit_spike = lambda([__s], 1-signum(__s)^2), __e))$
iif2if(__e) := block([opsubst : true], subst('iif = lambda([_cnd, __a, __b], %if(_cnd, __a , __b)), __e))$
if2iif(__e) := block([opsubst : true], subst('%if = lambda([_cnd, __a, __b], iif(_cnd, __a , __b)), __e))$
hstep2signum(__e):=block([opsubst : true], subst('hstep = lambda([_s], (1+signum(_s))/2), __e))$
signum2hstep(__e):=block([opsubst : true], subst('signum = lambda([_s], 2*hstep(_s)-1), __e))$


ifthen2if(__e):=block(
	[opsubst:true],
	subst("if"=lambda
	([[_v]],
	restofif(_v))
	,__e)
);

restofif(__l) := block(
	[_retval:0],
	if length(__l) # 0 and __l[1] # true then
		%if(__l[1], __l[2], restofif(rest(__l, 2)))
	elseif __l[1] = true then
		__l[2]
);


ifthen2iif(__e):=block(
	[opsubst:true],
	subst("if"=lambda
	([[_v]],
	restofiif(_v))
	,__e)
);

restofiif(__l) := block(
	[_retval:0],
	if length(__l) # 0 and __l[1] # true then
		iif(__l[1], __l[2], restofiif(rest(__l, 2)))
	elseif __l[1] = true then
		__l[2]
);


bffind_root(__expr, __f, __x, __low, __high, __digits):=
block
(
	[fpprec:fpprec+__digits+20, _side, _fb, _ft, _fn, _b, _t, _n, _retval:'unknown, _theaccuracy:10^-__digits],
	_b : bfloat(__low),
	_t : bfloat(__high),
    _fb : at(__expr,__x=_b),
    _ft : at(__expr,__x=_t),
	_fb : apply('ev, flatten(append([_fb], [__f]))),
	_ft : apply('ev, flatten(append([_ft], [__f]))),
	if sign(_ft) = sign(_fb) then
		print("Expression must not have same sign at both endpoints")
	else
	(
		_side : 0,  /* Falsi method */
		for i : 1 thru 10000 do (
			_n : (_fb*_t - _ft*_b) / (_fb - _ft),
			if (abs(_t-_b) < _theaccuracy * abs(_t+_b)) then
			(
				i : 10^6,
				fpprec:__digits,
				_retval : bfloat(_n)
			)
			else
			(
				_fn : at(__expr,__x=_n),
				_fn : apply('ev, flatten(append([_fn], [__f]))),
				if (_fn * _ft > 0) then
				(
				  _t : _n,
				  _ft : _fn,
				  if (_side = -1) then _fb : _fb / 2,
				  _side : -1
				)
				elseif (_fb * _fn > 0) then
				(
				  _b : _n,
				  _fb : _fn,
				  if (_side = 1) then _ft : _ft / 2,
				  _side : 1
				)
				else
				(
					i : 10^6,
					fpprec:__digits,
					_retval : bfloat(_n)
				)
			)
		)
	),
	_retval
)$

countvars(__e):=
block
(
	[inflag : true, _vars:[], _undupedvars:[],_retvals:[], _cnt, _i, _j],
	scanatoms(block([__z], lambda([__z], if not numberp(__z) and not member(__z, [%pi,%phi,%e,%i,'inf,'minf]) then _vars:cons(__z, _vars))), __e),
	_undupedvars:listify(setify(_vars)),
	for _i in _undupedvars do
	(
		_cnt : 0,
		for _j in _vars do
		(
			if is(_i=_j)=true then
			(
				_cnt : _cnt+1
			)
		),
		_retvals:cons([_i,_cnt],_retvals)
	),
	_retvals:sort(_retvals, lambda([__z1,__z2], if is(second(__z1) < second(__z2))=true then true else false))
)$


scanatoms(__f, __e):=
block(
	[inflag : true, _retval : true, _p, _done],
	if safe_op(__e) # false then
	(
		for _p in __e do (
			scanatoms(__f, _p)
		)
	)
	elseif safe_op(__e) # false and length(args(__e)) > 0 then
	(
		for _p in args(__e) do (
			scanatoms(__f, _p)
		)
	)
	elseif mapatom(__e) then
	(
		apply(__f,[__e])
	)
	else
	(
		apply(__f,[__e])
	),
	_done
)$

scanex(__f,__e):=
block
(
	[inflag : true, _p],
	if safe_op(__e) # false then  /* _p is an op() */
	(
		apply(__f, [__e]),
		for _p in __e do (
			if safe_op(_p) # false then  /* _p is an op() */
			(
				scanex(__f, _p),
				apply(__f, [_p])
			)
			else
				apply(__f, [_p])

		)
	)
	else
		apply(__f, [_p])

)$


max2iif(__e):=
block
(
	[inflag:true, opsubst:true, _retval : 0],
	_retval :
	subst('max = lambda([[_z]], apply('_listiifs1, _z)), __e)
)$

_listiifs1([_z]):=
block
(
	[inflag:true, opsubst:true, _len:length(_z), _iifs:[], _retval : 0],
	if length(_z) > 1 then
		_retval : simp_iif(iif(first(_z) > second(_z), apply('_listiifs1, list_remove(_z, 2, 1)), apply('_listiifs1, rest(_z))))
	else
		first(_z)
)$

min2iif(__e):=
block
(
	[inflag:true, opsubst:true, _retval : 0],
	_retval :
	subst('min = lambda([[_z]], apply('_listiifs2, _z)), __e)
)$

_listiifs2([_z]):=
block
(
	[inflag:true, opsubst:true, _len:length(_z), _iifs:[], _retval : 0],
	if length(_z) > 1 then
		_retval : simp_iif(iif(first(_z) < second(_z), apply('_listiifs2, list_remove(_z, 2, 1)), apply('_listiifs2, rest(_z))))
	else
		first(_z)
)$

maxmin2iif(__e):=max2iif(min2iif(__e));

12.2 Courbe de la fonction étudiée

(%i7) wxplot2d([g(t), f(t)], [t,a-b,3*b-2*a])$

(%t7)

12.3 Calcul des coefficients de Fourier

12.3.1 Calcul de la moyenne sur une période : a_0

(%i8) a0:1/T*integrate(g(t),t,a,b);

(%o8) 0

12.3.2 Calcul des coefficients a_n et b_n

(%i9) (declare(n,integer), n>0);

(%o9) n > 0

(%i10) an:2/T*integrate(g(t)*cos(n*%omega*t),t,a,b);
define(a(n),an);

(%o10) 0

(%o11) a (n) := 0

(%i12) bn:2/T*integrate(g(t)*sin(n*%omega*t),t,a,b);
define(b(n),bn);

(%o12) − 2 (−1)
n

n

(%o13) b (n) := −2 (−1)
n

n

N désigne le nombre de termes affichés. La fonction makelist permet de créer la liste des entiers
de 1 à N et map permet de caculer les images par la fonction a (ou b) des nombres de la liste
précédente.

(%i14) N:4$
map(’a, makelist(i,i,1,N));
map(’b, makelist(i,i,1,N));

(%o15) [0, 0, 0, 0]

(%o16) [2,−1,
2

3
,−1

2
]

12.4 Somme de Fourier partielle

On désigne par N le nombre de termes de la somme de série de Fourier partielle :

(%i17) Sf(t,N):=a0+sum(a(n)*cos(n*%omega*t)+b(n)*sin(n*%omega*t), n, 1, N);

(%o17) Sf (t,N) := a0 +
N∑
n=1

a (n) cos (nω t) + b (n) sin (nω t)

(%i18) wxplot2d([f(t),Sf(t,4)], [t,a-b,3*b-2*a])$

(%t18)

(%i19) wxplot2d([f(t),Sf(t,30)], [t,a-b,3*b-2*a])$

(%t19)

12.5 Spectre

Le spectre de f est donné par la suite (An), où An =
√
a2n + b2n, pour n ≥ 1.

(%i20) define(A(n), sqrt(a(n)^2+b(n)^2));

(%o20) A (n) :=
2

|n|

12.6 Valeur efficace et formule de Parseval

La valeur efficace est donnée par : f2eff =
1

T

∫ T+a

a

f2(t)4dt.

La formule de Parseval est : f2eff = a20 +
1

2

+∞∑
n=1

(a2n + b2n).

La valeur efficace est approchée par : P (N) = a20 +
1

2

N∑
n=1

(a2n + b2n).

(%i21) P(N):=a0^2+1/2*sum(a(n)^2+b(n)^2, n, 1, N);

(%o21) P (N) := a02 +
1

2

N∑
n=1

a (n)
2

+ b (n)
2

(%i22) N:10;
for n:1 thru N step 1 do print("P(",n,")=", float(P(n)), " ; ");

(%o22) 10P (1) = 2.0;P (2) = 2.5;P (3) = 2.722222222222222;P (4) = 2.847222222222222;P (5) =
2.927222222222222;P (6) = 2.982777777777778;P (7) = 3.02359410430839;P (8) = 3.05484410430839;P (9) =
3.079535462333082;P (10) = 3.099535462333082;

(%o23) done

(%i24) Peff:float(1/T*integrate(f(t)**2,t,a,b));

(%o24) 3.289868133696452

Erreur relative

(%i25) for n:1 thru N step 1 do print("err(",n,")=", float((Peff-P(n))/Peff), " ; ");

err(1) = 0.39207289814597; err(2) = 0.24009112268247; err(3) = 0.17254366692091; err(4) =
0.13454822305503; err(5) = 0.11023113898087; err(6) = 0.093344275040481; err(7) = 0.080937599492439; err(8) =
0.07143873852597; err(9) = 0.063933465663575; err(10) = 0.057854194645034;

(%o25) done

12.7 Utilisation du package piecewise

Package à utiliser pour les fonctions définie par morceaux sur une période.

(%i26) kill(all);

(%o0) done

(%i1) load("maxima-BTS_V_1/pw.mac");

(%o1) maxima−BTS_V_1/pw.mac

On suppose que la fonction g est donnée sur les intervalles \left[a_1; b_1\right], \ldtos, \left[a_r; b_r\right]
par les expressions g_1(x), \ldots, g_r(x). Dans Maxima et avec la bibliothèque pw, on peut définir
la fonction g directement par :

define(g(t), piecewise([a_1, g_1(t), b_1, a_2, g_2(t), b_2, \ldots, a_r, g_r(t), b_r,],t)) ;

(%i2) a:0;
b:4*%pi;
T:b-a;
%omega:2*%pi/T;
define(g(t), piecewise([a, %pi-t, %pi, 0, b/2, 1, b],t));
f(t):=g(t-(b-a)*floor((t-a)/(b-a)));
wxplot2d([f(t)], [t,a-b,3*b-2*a],[y,-0.5,3.5])$

(%o2) 0

(%o3) 4π

(%o4) 4π

(%o5)
1

2

(%o6) g (t) :=
(π − t) (signum (t)− signum (t− π))

2
+

signum (t− 2π)− signum (t− 4π)

2

(%o7) f (t) := g

(
t− (b− a) floor

(
t− a
b− a

))

(%t8)

La fonction pwint est la fonction integrate utilisable sur les fonctions définies par pw, c’est l’intégrale
d’une fonction définie par morceaux.

(%i9) a0:radcan(1/T*pwint(g(t), t, a, b));

(%o9)
π + 4

8

(%i10) declare(n, integer);
assume(n>0);

(%o10) done

(%o11) [n > 0]

La fonction radcan, sert à simplifier l’expression obtenue (simplification canonique).

(%i12) an:radcan(2/T*pwint(g(t)*cos(n*%omega*t), t, a, b));
define(a(n),an);

(%o12) −
2 cos

(
π n
2

)
− 2

π n2

(%o13) a (n) := −
2 cos

(
π n
2

)
− 2

π n2

(%i14) bn:radcan(2/T*pwint(g(t)*sin(n*%omega*t), t, a, b));
define(b(n),bn);

(%o14) −
2 sin

(
π n
2

)
− n (−1)

n
+ (1− π) n

π n2

(%o15) b (n) := −
2 sin

(
π n
2

)
− n (−1)

n
+ (1− π) n

π n2

(%i16) N:4$
map(’a, makelist(i,i,1,N));
map(’b, makelist(i,i,1,N));

(%o17) [
2

π
,

1

π
,

2

9π
, 0]

(%o18) [−4− π
π

,−2 (1− π)− 2

4π
,−3 (1− π) + 1

9π
,−4 (1− π)− 4

16π
]

(%i19) Sf(t,N):=a0+sum(a(n)*cos(n*%omega*t)+b(n)*sin(n*%omega*t), n, 1, N);

(%o19) Sf (t,N) := a0 +

N∑
n=1

a (n) cos (nω t) + b (n) sin (nω t)

(%i20) wxplot2d([f(t),Sf(t,5)], [t,a-b,3*b-2*a])$

(%t20)

(%i21) wxplot2d([f(t),Sf(t,30)], [t,a-b,3*b-2*a])$

(%t21)



Chapitre 13

Transformée de Laplace

13.1 Définition de l’échelon unité

Échelon unité : v(t) = 0 si t < 0 et 1 si t > 0 ;

Dans Maxima c’est la fonction u suivante qui est définie : u(t) = 0 si t <= 0 et 1 si t > 0 ;

Dans la pratique en mathématiques, c’est la fonction v que l’on utilise mais avec maxima c’est avec
la fonction u que cela fonctionne correctement !

(%i1) u(t):=unit_step(t);
u(-1);
u(0);
u(1);

(%o1) u (t) := unit_step (t)

(%o2) 0

(%o3) 0

(%o4) 1

La fonction v peut être définie de la façon suivante :

(%i5) v(t):=unit_step(t)+kron_delta(t,0);
v(-1);
v(0);
v(1);

(%o5) v (t) := unit_step (t) + kron_delta (t, 0)

(%o6) 0

(%o7) 1

(%o8) 1

Les calculs sur la transformée de Laplace ne changent pas quel que soit le choix d’une des deux
définitions précédentes.

13.2 Calculs de transformées de Laplace

Menu/Calculs/Transformée de Laplace...

85

(%i9) laplace(u(t), t, p);

(%o9)
1

p

(%i10) laplace(t*u(t), t, p);

(%o10)
1

p2

Ci-dessous, Γ(n+ 1) désigne n!.

(%i11) laplace(t^n*u(t), t, p);

Isn+ 1positive, negative, orzero?positive;

(%o11) γ (n+ 1) p−n−1

Retard u(t− 1)

(%i12) laplace(u(t-tau), t, p);

(%o12)
e−p τ

p

Exponentielle

(%i13) laplace(exp(-a*t), t, p);

(%o13)
1

p+ a

Fonctions trigonométriques

(%i14) laplace(cos(omega*t)*u(t), t, p);

(%o14)
p

p2 + ω2

(%i15) laplace(sin(omega*t)*u(t), t, p);

(%o15)
ω

p2 + ω2

Propriétés : les deux premières ne sont par reconnues...

(%i16) laplace(f(alpha*t)*u(alpha*t), t, p);

(%o16) laplace (f (α t) unit_step (α t) , t, p)

(%i17) laplace(f(t-tau), t, p);

(%o17) laplace (f (t− τ) , t, p)

(%i18) laplace(exp(-a*t)*f(t), t, p);

Isp+ apositive, negative, orzero?positive;

(%o18) laplace (f (t) , t, p)|p=p+a

(%i19) laplace(diff(f(t), t),t,p);

(%o19) p laplace (f (t) , t, p)− f (0)

(%i20) laplace(diff(f(t), t, 2),t,p);

(%o20) − d

d t
f (t)

∣∣∣∣
t=0

+ p2 laplace (f (t) , t, p)− f (0) p

(%i21) laplace(-t*f(t), t, p);

(%o21)
d

d p
laplace (f (t) , t, p)

(%i22) laplace(integrate(f(s), s, 0, t), t, p);

(%o22)
laplace (f (t) , t, p)

p

13.3 Calculs de transformées inverse de Laplace

Menu/Calculs/Transformée inverse de Laplace... Attention, il faut multiplier les résultats par u(t).

(%i23) ilt(1/p, p, t);

(%o23) 1

Si le dénominateur est du second degré avec un discriminant strictement positif :

(%i24) ilt((p+1)/(p^2+4*p-1), p, t);

(%o24) e−2 t
(

cosh
(√

5 t
)
−

sinh
(√

5 t
)

√
5

)
Des fonctions trigonométriques hyperboliques ! Pour éviter cela, on utilise exponentialize :

(%i25) exponentialize(ilt((p+1)/(p^2+4*p-1), p, t));

(%o25) e−2 t
(
e
√
5 t + e−

√
5 t

2
− e
√
5 t − e−

√
5 t

2
√

5

)
On peut aussi développer !

(%i26) expand(exponentialize(ilt((p+1)/(p^2+4*p-1), p, t)));

(%o26) − e
√
5 t−2 t

2
√

5
+
e
√
5 t−2 t

2
+
e−
√
5 t−2 t

2
√

5
+
e−
√
5 t−2 t

2

Si le dénominateur est du second degré avec un discriminant nul :

(%i27) ilt((p+1)/(p^2+2*p+1), p, t);

(%o27) e−t

Rien à modifier ! Si le dénominateur est du second degré avec un discriminant strictement négatif :

(%i28) ilt((p+1)/(p^2+p+1), p, t);

(%o28) e−
t
2

 sin
(√

3 t
2

)
√

3
+ cos

(√
3 t

2

)

Attention : "ilt" (inverse laplace transform) ne fonctionne que pour les fonctions rationnelles pour
lesquelles le dénominateur est factorisable en produits de facteurs linéaires et quadratiques (dont
on sait caluler la valeur exacte des racines). Cela pose des problèmes si l’on veut travailler avec des
retards. Il faut dans ce cas utiliser un autre logiciel ou trouver une autre solution !

13.4 Résolutions d’équations différentielles avec la transfor-
mée de Laplace

– Résolution directe : Menu/Equations/Résoudre une équation différentielle avec Laplace...

(%i29) desolve([’diff(y(t), t, 2) + 5*’diff(y(t), t) + 4*y(t) = u(t)],[y(t)]);

(%o29) y (t) = −
e−4 t

(
4
(
d
d t y (t)

∣∣
t=0

)
+ 4 y (0)− 1

)
12

+
e−t

(
d
d t y (t)

∣∣
t=0

+ 4 y (0)− 1
)

3
+

1

4

Contrairement à ode2 les paramètres restants sont les valeurs de y(0) et y’(0) et non les deux réels
quelconques notés %k1 et %k2 dans maxima. Pour faciliter la lecture des résultats on pose les
conditions initiales f(0) = y0 et f’(0) = yp0 :

(%i30) atvalue(y(t), t=0, y0);
atvalue(’diff(y(t), t), t= 0, yp0);

(%o30) y0

(%o31) yp0

On affiche le changement :

(%i32) ’’%i19;

(%o32) p laplace (f (t) , t, p)− f (0)

On développe :

(%i33) expand(%o22);

(%o33)
laplace (f (t) , t, p)

p

Avec les conditions initiales : y(0)=1 et y’(0)=2 :

(%i34) subst(1,y0,%o23);

(%o34) 1

(%i35) subst(2,yp0,%o24);

(%o35) e−2 t
(

cosh
(√

5 t
)
−

sinh
(√

5 t
)

√
5

)
– Par étapes :

(%i36) equation: ’diff(y(t), t, 2) + 5*’diff(y(t), t) + 4*y(t) = u(t);

(%o36)
d2

d t2
y (t) + 5

(
d

d t
y (t)

)
+ 4 y (t) = unit_step (t)

(%i37) equationtransformee:laplace(equation,t,p);

(%o37) −yp0−p y0+5 (p laplace (y (t) , t, p)− y0)+p2 laplace (y (t) , t, p)+4 laplace (y (t) , t, p) =
1

p

On impose des conditions initiales :

(%i38) atvalue(y(t), t=0, 1);
atvalue(’diff(y(t), t), t= 0, 2);
equationtransformee:laplace(equation,t,p);

(%o38) 1

(%o39) 2

(%o40) 5 (p laplace (y (t) , t, p)− 1) + p2 laplace (y (t) , t, p) + 4 laplace (y (t) , t, p)− p− 2 =
1

p

On résout l’équation linéaire :

(%i41) equationsolution: solve(equationtransformee, ’laplace(y(t), t, p));

(%o41) [laplace (y (t) , t, p) =
p2 + 7 p+ 1

p3 + 5 p2 + 4 p
]

On donne directement l’original. Si map et lambda paraissent obscurs, on peut faire un copier
coller !

(%i42) map( lambda( [eq], ilt(eq, p, t)), equationsolution);

(%o42) [y (t) =
5 e−t

3
− 11 e−4 t

12
+

1

4
]

On peut aussi décomposer le calcul en passant par la décomposition en éléments simples.

(%i43) map( lambda( [expression], partfrac(expression, p)), equationsolution);

(%o43) [laplace (y (t) , t, p) = − 11

12 (p+ 4)
+

5

3 (p+ 1)
+

1

4 p
]

(%i44) map( lambda( [expression], ilt(expression, p, t)), %);

(%o44) [y (t) =
5 e−t

3
− 11 e−4 t

12
+

1

4
]

L’ordre des termes a été modifié !

Chapitre 14

Transformée en z

Pour ce chapitre, il faut utiliser deux bibliothèques, pour les récupérer, cliquer deux fois sur les
liens ci-contre : lien 1 et lien 2 .

Enregistrez ces fichier dans le répertoire de votre choix et n’oubliez pas de donner le chemin complet
lord de l’utilisation :

load("C:/chemin-complet/z-transform.mac");

load("C:/chemin-complet/supplement-z-transform.mac");

14.1 Introduction

Définition des fonctions tz (transformée en z), tiz (transformée inverse en z) tiz retourne une liste,
les premiers termes sont donnés lorsqu’ils sont nuls, le dernier terme de la liste (disons d’ordre r
dans la liste) donne l’expression du signal discret en fonction de n pour n plus grand ou égal à r.
On peut obtenir ce terme générique directement avec la fonction tizgen. En utilisant la fonction
plotliste, on peut représenter le signal discret.

(%i1) load("z-transform.mac")$
load("supplement-z-transform.mac")$

14.2 Transformée en z

Des exemples :

Dirac (impulsion unité discrète) et sa trasformée

(%i3) dirac(n):=kron_delta(n,0)$
tz(dirac(n));

(%o4) 1

Dirac retardé de k et sa transformée

(%i5) d(n,k):=kron_delta(n,k)$
tz(d(n,k));

(%o6)
1

zk

/* z-transform code
 * Copyright 2007 by Robert Dodier
 * I release this work under terms of GNU General Public License
 *
 * Summary. Z-transforms for various special cases are implemented as
 * pattern-matching rules. Given an expression like z_transform(foo(n, z), n, z),
 * for most rules it's necessary to have the variables n and z in hand before
 * looking at foo(n, z). However, the Maxima pattern matcher does not have
 * backtracking, so one rule (r0 below) captures n and z, then all other
 * rules are applied.
 */

put ('z_transform, true, 'present);

apply_z_transform (e) := apply1 (e, r0);

matchdeclare ([nn, zz], symbolp);
matchdeclare (aa, all);

defrule (r0,
    z_transform (aa, nn, zz),
    block ([nn% : nn, zz% : zz],
        apply1
           (z_transform (aa, nn, zz),
            r913_1a, r913_1b, r913_2a, r913_2b, r913_3a, r913_3b,
            r913_4, r913_5, r913_6, r913_7, r913_10, r913_12, r914_4,
            r914_6, /* r914_9, */ r914_10, r914_11a, r914_11b, r914_12,
            r914_13, r914_14, r914_15a, r914_16)));

/* Some specific transforms.
 * Table 9.1.3 at: http://math.fullerton.edu/mathews/c2003/ZTransformIntroMod.html
 */

/* (1) delta[n] --> 1
 * (have to try kron_delta both ways ... sigh)
 */

simp : false;

defrule (r913_1a,
    z_transform (kron_delta (nn%, aa), nn%, zz%),
    zz^(- aa));

defrule (r913_1b,
    z_transform (kron_delta (aa, nn%), nn%, zz%),
    zz^(- aa));

simp : true;

/* (2) u[n] --> z/(z - 1) */

defrule (r913_2a,
    z_transform (1, nn%, zz%),
    zz/(zz - 1));

defrule (r913_2b,
    z_transform (unit_step (nn%), nn%, zz%),
    zz/(zz - 1));

/* (3) b^n --> z/(z - b) */

matchdeclare (bb, freeof (nn%, zz%));

defrule (r913_3a,
    z_transform (bb^nn%, nn%, zz%),
    zz/(zz - bb));

defrule (r913_3b,
    z_transform (1/(bb^nn%), nn%, zz%),
    zz/(zz - 1/bb));

/* (4) b^(n - 1) * u[n - 1] --> 1/(z - b) */

defrule (r913_4,
    z_transform (bb^(nn% - 1) * unit_step (nn% - 1), nn%, zz%),
    1 / (zz - bb));

/* (5) e^(a*n) --> z/(z - e^a) */

matchdeclare (aa, lambda ([e], e # 0 and freeof (nn%, zz%, e)));

defrule (r913_5,
    z_transform (exp (aa * nn%), nn%, zz%),
    zz / (zz - exp (aa)));

/* (6) n --> z/(z - 1)^2 */

defrule (r913_6,
    z_transform (nn%, nn%, zz%),
    zz / (zz - 1)^2);

/* (7) n^2 --> z*(z + 1)/(z - 1)^3 */

defrule (r913_7,
    z_transform (nn%^2, nn%, zz%),
    zz*(zz + 1) / (zz - 1)^3);

/* (8) b^n*n --> b*z/(z - b)^2
 * via (6) + frequency scaling
 */

/* (9) e^(a*n)*n --> z*e^a/(z - e^a)^2
 * via (6) + complex translation
 */

/* (10) sin(a*n) --> sin(a)*z/(z^2 - 2*cos(a)*z + 1) */

defrule (r913_10,
    z_transform (sin (aa*nn%), nn%, zz%),
    sin(aa)*zz / (zz^2 - 2*cos(aa)*zz + 1));

/* (11) b^n*sin(a*n) --> sin(a)*b*z/(z^2 - 2*cos(a)*b*z + b^2)
 * via (10) + frequency scaling
 */

/* (12) cos(a*n) --> z*(z - cos(a))/(z^2 - 2*cos(a)*z + 1) */

defrule (r913_12,
    z_transform (cos (aa*nn%), nn%, zz%),
    zz*(zz - cos(aa)) / (zz^2 - 2*cos(aa)*zz + 1));

/* (13) b^n*cos(a*n) --> z*(z - b*cos(a))/(z^2 - 2*cos(a)*b*z + b^2)
 * via (11) + frequency scaling
 */


/* General properties.
 * Table 9.1.4 at: http://math.fullerton.edu/mathews/c2003/ZTransformIntroMod.html
 */

/* (4) u[n - m] --> z^(1 - m)/(z - 1)
 * (delayed unit step)
 */

matchdeclare ([aa, mm], all);

defrule (r914_4,
    z_transform (unit_step (nn% - mm), nn%, zz%),
    zz^(1 - mm) / (zz - 1));
    
/* (5) x[n - 1]*u[n - 1] --> (1/z)*X(z)
 * via (6) w/ m = 1
 */

/* (6) x[n - m]*u[n - m] --> z^(-m)*X(z)
 * (time delayed shift)
 */

defrule (r914_6,
    z_transform (aa * unit_step (nn% - mm), nn%, zz%),
    zz^(-mm) * z_transform (subst (nn + mm, nn, aa), nn, zz));

/* (7) x[n + 1] --> z*(X(z) - x[0])
 * (8) x[n + 2] --> z^2*(X(z) - x[0] - x[1]*z^(-1))
 * via (9) w/ m = 1 and m = 2 respectively
 */

/* (9) x[n + m] --> z^m*(X(z) - sum(x[i]*z^(-i), i, 0, m - 1))
 * (time forward)
 */

/* HMM, NOT SURE HOW TO DO (9) ... FOLLOWING STUFF IS BROKEN */

matchdeclare (mm, lambda ([e], integerp(e) and e > 0));
defmatch (n_plus_m, nn% + mm);


matchdeclare (xxnpm, lambda ([e], n_plus_m (e) # false));


defrule (r914_9,
   z_transform (aa * unit_step (nn% + mm), nn%, zz%),
   zz^(mm+1) * (z_transform (subst (nn - mm, nn, ss), nn, zz)-1)
   );

 
/* (10) e^(a*n)*x[n] --> X(z*e^(-a))
 * (complex translation)
 */

matchdeclare (nz, lambda ([e], e # 0 and freeof (nn%, zz%, e)));

defrule (r914_10,
    z_transform (exp (nz * nn%) * bb, nn%, zz%),
    'subst (zz/exp(nz), zz, z_transform (bb, nn, zz)));

/* (11) b^n*x[n] --> X(z/b)
 * (frequency scaling)
 */

matchdeclare (xx, all);
matchdeclare (bb, freeof (nn%, zz%));

defrule (r914_11a,
    z_transform (bb^nn% * xx, nn%, zz%),
    'subst (zz/bb, zz, z_transform (xx, nn, zz)));

defrule (r914_11b,
    z_transform (1/(bb^nn%) * xx, nn%, zz%),
    'subst (zz*bb, zz, z_transform (xx, nn, zz)));

/* (12) n*x[n] --> -z X'(z)
 * (differentiation)
 */

matchdeclare (aa, all);
matchdeclare (kk, lambda ([e], integerp(e) and e > 0));

defrule (r914_12,
    z_transform (aa*nn%^kk, nn%, zz%),
    block ([ee : - zz * 'diff (z_transform (aa, nn, zz), zz)],
        for i:2 thru kk do ee : - zz * 'diff (ee, zz), ee));

/* (13) (1/n)*x[n] --> - \int X(z)/z dz
 * (integration)
 */

matchdeclare (uu, lambda ([e], not atom(e) and op(e) = "/" and member (nn%, second(e))));

defrule (r914_13,
    z_transform (uu, nn%, zz%),
    'integrate (zz^-1 * z_transform (aa, nn, zz), zz));

/* (14) 1/(n + m)*x[n] --> - z^(-m) * \int X(z)/z^(m + 1) dz
 * (integration shift)
 */

matchdeclare (mm, lambda ([e], integerp(e) and e > 0));

defrule (r914_14,
    z_transform (aa/(nn% + mm), nn%, zz%),
    - zz^(-mm) * 'integrate (z^-(mm + 1) * z_transform (aa, nn, zz), zz));

/* (15) x[n] (star) y[n] = \sum_{i=0}^n x[i]*y[n - i] --> X(z)*Y(z)
 * (discrete time convolution)
 */

matchdeclare (cc, lambda ([e], not atom(e) and op(e) = 'convolution));

defrule (r914_15a,
    z_transform (cc, nn%, zz%),
    block ([a : args(cc)], product (z_transform (a[i], nn, zz), i, 1 , length(a))));

/* ANOTHER RULE R914_15B FOR EXPLICIT 'SUM EXPRESSION WOULD BE NICE ... */

/* (16) \sum_{i=0}^n x[i] --> z/(z - 1)*X(z)
 * (convolution with y[n] = 1)
 */

matchdeclare (ii, symbolp);

simp : false;

defrule (r914_16,
    z_transform ('sum (aa, ii, 0, nn%), nn%, zz%),
    zz/(zz - 1) * z_transform (subst (nn, ii, aa), nn, zz));

simp : true;

/* ((17) & (18) -- not transform pairs) */

/* (put linearity declaration last, otherwise messes up rules) */

/* (1) addition
 * (2) constant multiple
 * (3) linearity
 */

declare (z_transform, linear);




allOpsPriv(expression, opList) :=
 block ( [x, args, newList],
        if atom(expression)
           then opList
           else
             (x:    op(expression),
              args: args(expression),
              newList: if member (x, opList)
                          then opList
                          else cons(x, opList),
              for arg in args do
                newList: allOpsPriv(arg, newList),
              newList
             ) 
        );

allOps(expression):=
 block( [ ],
        allOpsPriv (expression, [])
       );

isEquation(expression) := block( [ ],
if member ("=",allOps(expression))
    then true
    else false
	);


tz(f):=block([],
	if isEquation(f)
		then apply_z_transform(z_transform(lhs(f), n, z))=apply_z_transform(z_transform(rhs(f), n, z))
		else apply_z_transform(z_transform(f, n, z))
);

tizpole(f,polea,n)::=buildq ([f,n,z,polea], residue(f*z^(n-1),z,polea));

zeros(f)::=buildq ([f,z], block([],
Listesolve:solve(f,z),
Listezeros:[],
longueurListesolve:length(Listesolve),
for n:1 thru longueurListesolve step 1 do Listezeros:append(Listezeros,[rhs(Listesolve[n])]),
Listezeros
));

poles(f)::=buildq ([f,z], block([],
fp:diff(f,z),
Listea:zeros(f/fp),
Listeb:zeros(f),
longueurListea:length(Listea),
Liste:[],
for n:1 thru longueurListea step 1 do 
block([],
if member(Listea[n],Listeb)
then none
else Liste:append(Liste,[Listea[n]])
),
Liste
));

ordrepolezero(exprf)::=buildq ([exprf,z,exprfp,g], block([],
define(exprfp(z),diff(exprf,z)),
define(g(z),ratsimp(-exprfp(z)/exprf*z)),
g(0)
));

tizgen(f,n):=lsum(residue(f*z^(n-1),z,b),b,poles(f));

tiz(exprf)::=buildq ([exprf,z], block([],
k:max(ordrepolezero(exprf),0),
Liste:[limit(exprf,z,inf)],
for l:1 thru k step 1 do Liste:append(Liste,[tizgen(exprf,l)]),
Liste:append(Liste,[tizgen(exprf,n)]),
Liste
));

plotliste(liste, fin):=wxplot2d([discrete, makelist(k,k,0,fin),append(makelist(liste[k],k, 1,length(liste)-1),makelist(ev(liste[length(liste)],n=k),k,length(liste)-1,fin))],[style, points]);

tzliste(liste):=sum(liste[k]/(z^(k-1)),k,1, length(liste)-1)+1/z^(length(liste)-1)*apply_z_transform(z_transform(subst(n+length(liste)-1, n, liste[length(liste)]),n,z));

Échelon unité discret

(%i7) u(n):=unit_step(n)$
tz(u(n));

(%o8)
z

z − 1

Rampe unité causale discrète : n

(%i9) tz(n*u(n));
factor(ev(’’(tz(n*u(n))),diff));
tz(n);

(%o9) − z
(
d

d z

z

z − 1

)
(%o10)

z

(z − 1)
2

(%o11)
z

(z − 1)
2

Carré

(%i12) tz(n^2*u(n));
factor(ev(’’(tz(n^2*u(n))),diff));
tz(n^2);

(%o12) − z
(
d

d z

(
−z

(
d

d z

z

z − 1

)))
(%o13)

z (z + 1)

(z − 1)
3

(%o14)
z (z + 1)

(z − 1)
3

Exponentiel

(%i15) tz(a^n*u(n));
factor(ev(’’(tz(a^n*u(n))),nouns));
tz(a^n);

(%o15) substitute

(
z

a
, z,

z

z − 1

)
(%o16)

z

z − a

(%o17)
z

z − a

(%i18) tz(b^(n-1)*u(n-1));

(%o18)
1

z − b

(%i19) tz(exp(a*n));

(%o19)
z

z − ea

Trigonométrie

(%i20) tz(sin(a*n));
tz(cos(a*n));

(%o20)
sin (a) z

z2 − 2 cos (a) z + 1

(%o21)
z (z − cos (a))

z2 − 2 cos (a) z + 1

(%i22) factor(ev(’’(tz(b^n*cos(a*n))),nouns));
factor(ev(’’(tz(b^n*sin(a*n))),nouns));

(%o22)
z (z − cos (a) b)

z2 − 2 cos (a) b z + b2

(%o23)
sin (a) b z

z2 − 2 cos (a) b z + b2

Des propriétés Linéarité

(%i24) tz(lambda*x(n)+mu*y(n));

(%o24) z_transform (x (n) , n, z) lambda+ µ z_transform (y (n) , n, z)

Retard

(%i25) tz(x(n-2)*u(n-2));
y(n):=n^2*u(n);
factor(ev(’’(tz(y(n-2))),diff));

(%o25)
z_transform (x (n) , n, z)

z2

(%o26) y (n) := n2 u (n)

(%o27)
z + 1

(z − 1)
3
z

L’avance ne fonctionne pas encore !

(%i28) tz(x(n+3)*unit_step(n+3));
y(n):=n^2*u(n);
ev(’’(tz(y(n+2))), diff);
tz(expand((n+2)^2));

(%o28) z_transform (x (n) , n, z) z3

(%o29) y (n) := n2 u (n)

(%o30) − z3
(
−z

(
2 z

(z − 1)
3 −

2

(z − 1)
2

)
+

z

(z − 1)
2 −

1

z − 1

)

(%o31)
z (z + 1)

(z − 1)
3 +

4 z

z − 1
+

4 z

(z − 1)
2

Multiplication par a^n

(%i32) tz(a^n*x(n));
y(n):=n^2*u(n);
tz(a^n*y(n));
factor(ev(’’(tz(a^n*y(n))),nouns));

(%o32) substitute
(z
a
, z, z_transform (x (n) , n, z)

)

(%o33) y (n) := n2 u (n)

(%o34) substitute

(
z

a
, z,−z

(
d

d z

(
−z

(
d

d z

z

z − 1

))))
(%o35)

a z (z + a)

(z − a)
3

14.3 Transformée en z inverse

Le premier terme donne la valeur du signal discret pour n=0. Le dernier terme est l’expression du
terme général du signal.

(%i36) tiz(z/(z-1));

(%o36) [1, 1]

(%i37) tiz(z/(z-1)^2);

(%o37) [0, n]

(%i38) tiz((z*(z+1))/(z-1)^3);

(%o38) [0, n2]

(%i39) tiz(z/(z-b));

(%o39) [1, bn]

(%i40) tiz(z/(z-%e^a));

(%o40) [1, ean]

(%i41) tiz(z^(6)/(z-1)^2);

(%o41) [∞, n+ 5]

(%i42) define(f(z),z^(-6)/(z-1)^2);
signal:tiz(f(z));
plotliste(signal,15);

(%o42) f (z) :=
1

(z − 1)
2
z6

(%o43) [0, 0, 0, 0, 0, 0, 0, n− 7]

(%t44)

(%o44)

14.4 Transformée en z à partir d’une liste

(%i45) tzliste([1.1,1.2,1.3,1.4,1.5*n]);

(%o45)
1.5

(
4 z
z−1 + z

(z−1)2

)
z4

+
1.2

z
+

1.3

z2
+

1.4

z3
+ 1.1

(%i46) tzliste([0,0,n-2]);

(%o46)
1

(z − 1)
2
z

(%i47) factor(tzliste([n+1]));

(%o47)
z2

(z − 1)
2

14.5 Exemples d’utilisations

(%i48) tz(n^2+3*n+2);

(%o48)
z (z + 1)

(z − 1)
3 +

2 z

z − 1
+

3 z

(z − 1)
2

(%i49) factor(ev(’’(tz(n*2^n)),diff, nouns));

(%o49)
2 z

(z − 2)
2

(%i50) factor(tz(n-3));
factor(tzliste([0, 0, 0, n-3]));

(%o50) − z (3 z − 4)

(z − 1)
2

(%o51)
1

(z − 1)
2
z2

(%i52) factor(tz(expand((n+2)^2)));
factor(z^3*(z+1)/(z-1)^3-z);

(%o52)
z
(
4 z2 − 3 z + 1

)
(z − 1)

3

(%o53)
z
(
4 z2 − 3 z + 1

)
(z − 1)

3

(%i54) tiz(2*z/((z-1)*(z-3)));

(%o54) [0, 3n − 1]

(%i55) tiz(3/(z-2));

(%o55) [0, 3 2n−1]

(%i56) residue(3/(z*(z-2)),z,2);

(%o56)
3

2

(%i57) tiz(z/(z-1));

(%o57) [1, 1]

(%i58) tiz(1/(z-1));

(%o58) [0, 1]

(%i59) tiz(1/(z^5*(z-1)));

(%o59) [0, 0, 0, 0, 0, 0, 1]

(%i60) residue(1/(z*(z-1)),z,1);

(%o60) 1

(%i61) tiz((2*z^2)/(11*z^2-21*z+10));

(%o61) [
2

11
, 2− 2 11−n−1 10n+1]

14.6 Résolution d’équations récurrentes avec la transformée
en z

(%i62) kill(x, y);

(%o62) done

(%i63) equation : 11*y(n)-10*y(n-1)*unit_step(n-1) = 2*unit_step(n);

(%o63) 11 y (n)− 10 unit_step (n− 1) y (n− 1) = 2 unit_step (n)

(%i64) equationtransformee:tz(equation);

(%o64) 11 z_transform (y (n) , n, z)− 10 z_transform (y (n) , n, z)

z
=

2 z

z − 1

(%i65) solve(equationtransformee, ’’tz(y(n)))[1];

(%o65) z_transform (y (n) , n, z) =
2 z2

11 z2 − 21 z + 10

(%i66) tiz(rhs(%));

(%o66) [
2

11
, 2− 2 11−n−1 10n+1]

